

Dependently Typed Metaprogramming (in Agda)

Conor McBride

August 20, 2013

Introduction

If you have never met a metaprogram in a dependently typed programming language like Agda [Norell, 2008], then prepare to be underwhelmed. Once we have types which can depend computationally upon first class values, metaprograms just become ordinary programs manipulating and interpreting data which happen to stand for types and operations.

This course, developed in the summer of 2013, explores methods of metaprogramming in the dependently typed setting. I happen to be using Agda to deliver this material, but the ideas transfer to any setting with enough dependent types. It would certainly be worth trying to repeat these experiments in Idris, or in Coq, or in Haskell, or in your own dependently typed language, or maybe one day in mine.

Chapter 1

Vectors and Normal Functors

It might be easy to mistake this chapter for a bland introduction to dependently typed programming based on the yawning-already example of lists indexed by their length, known to their friends as *vectors*, but in fact, vectors offer us a way to start analysing data structures into ‘shape and contents’. Indeed, the typical motivation for introducing vectors is exactly to allow types to express shape invariants.

1.1 Zipping Lists of Compatible Shape

Let us remind ourselves of the situation with ordinary *lists*, which we may define in Agda as follows:

```
data List (X : Set) : Set where
  ⟨⟩      : List X
  –, _ : X → List X → List X
infixr 4 –, _
```

The classic operation which morally involves a shape invariant is *zip*, taking two lists, one of *S*s, the other of *T*s, and yielding a list of pairs in the product $S \times T$ formed from elements *in corresponding positions*. The trouble, of course, is ensuring that positions correspond.

```
zip : {S T : Set} → List S → List T → List (S × T)
zip ⟨⟩      ⟨⟩      = ⟨⟩
zip (s, ss) (t, ts) = (s, t), zip ss ts
zip _      _      = ⟨⟩ -- a dummy value, for cases we should not reach
```

Agda has a very simple lexer and very few special characters. To a first approximation, `(){};` stand alone and everything else must be delimited with whitespace.

The braces indicate that *S* and *T* are *implicit arguments*. Agda will try to infer them unless we override manually.

Overloading Constructors Note that I have used ‘,’ both for tuple pairing and as list ‘cons’. Agda permits the overloading of constructors, using type information to disambiguate them. Of course, just because overloading is permitted, that does not make it compulsory, so you may deduce that I have overloaded deliberately. As data structures in the memory of a computer, I think of pairing and consing as the same, and I do not expect data to tell me what they mean. I see types as an external rationalisation imposed upon the raw stuff of computation, to help us check that it makes sense (for multiple possible notions of sense) and indeed to infer details (in accordance with notions of sense). Those of you who have grown used to thinking of type annotations as glorified comments will need to retrain your minds to pay attention to them.

Our `zip` function imposes a ‘garbage in? garbage out!’ deal, but logically, we might want to ensure the obverse: if we supply meaningful input, we want to be sure of meaningful output. But what is meaningful input? Lists the same length! Locally, we have a *relative* notion of meaningfulness. What is meaningful output? We could say that if the inputs were the same length, we expect output of that length. How shall we express this property? We could externalise it in some suitable program logic, first explaining what ‘length’ is.

The number of c’s in `suc` is a long standing area of open warfare.

Agda users tend to use lowercase-`vs`-uppercase to distinguish things in `Sets` from things which are or manipulate `Sets`.

The pragmas let you use Arabic numerals.

```
data ℕ : Set where
  zero : ℕ
  suc  : ℕ → ℕ
{-# BUILTIN NATURAL Nat #-}
{-# BUILTIN ZERO zero #-}
{-# BUILTIN SUC suc #-}

length : {X : Set} → List X → ℕ
length ⟨⟩      = zero
length (x, xs) = suc (length xs)
```

Informally,¹ we might state and prove something like

$$\forall ss, ts. \text{length } ss = \text{length } ts \Rightarrow \text{length } (\text{zip } ss \ ts) = \text{length } ss$$

by structural induction [Burstall, 1969] on `ss`, say. Of course, we could just as well have concluded that $\text{length } (\text{zip } ss \ ts) = \text{length } ts$, and if we carry on zipping, we shall accumulate a multitude of expressions known to denote the same number.

Matters get worse if we try to work with matrices as lists of lists (a matrix is a column of rows, say). How do we express rectangularity? Can we define a function to compute the dimensions of a matrix? Do we want to? What happens in degenerate cases? Given m, n , we might at least say that the outer list has length m and that all the inner lists have length n . Talking about matrices gets easier if we imagine that the dimensions are *prescribed*—to be checked, not measured.

1.2 Vectors

Dependent types allow us to *internalize* length invariants in lists, yielding *vectors*. The index describes the shape of the list, thus offers no real choice of constructors.

```
data Vec (X : Set) : ℕ → Set where
  ⟨⟩ : Vec X zero
  →, - : {n : ℕ} → X → Vec X n → Vec X (suc n)
```

Parameters and indices. In the above definition, the element type is abstracted uniformly as X across the whole thing. The definition could be instantiated to any particular set X and still make sense, so we say that X is a *parameter* of the definition. Meanwhile, `Vec`’s second argument varies in each of the three places it is instantiated, so that we are really making a mutually inductive definition of the vectors at every possible length, so we say that the length is an *index*. In an Agda `data` declaration head, arguments left of `:` (X here) scope over all constructor declarations and must be used uniformly in constructor return types, so it is sensible to put parameters left of `:`. However, as we shall see, such arguments may be

¹by which I mean, not to a computer

freely instantiated in *recursive* positions, so we should not presume that they are necessarily parameters.

Let us now develop `zip` for vectors, stating the length invariant in the type.

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ss ts = ?
```

The length argument and the two element types are marked implicit by default, as indicated by the `{. .}` after the **forall**. We write a left-hand-side naming the explicit inputs, which we declare equal to an unknown `?`. Loading the file with `[C - c C - l]`, we find that Agda checks the unfinished program, turning the `?` into labelled braces,

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ss ts = { }0
```

and tells us, in the information window,

```
?0 : Vec (.S × .T) .n
```

that the type of the ‘hole’ corresponds to the return type we wrote. The dots before `S`, `T`, and `n` indicate that these variables exist behind the scenes, but have not been brought into scope by anything in the program text: Agda can refer to them, but we cannot.

If we click between the braces to select that hole, and issue keystroke `[C - c C - ,]`, we will gain more information about the goal:

```
Goal : Vec (Σ .S (λ _ . T)) .n
```

```
ts   : Vec .T .n
ss   : Vec .S .n
.T   : Set
.S   : Set
.n   : ℕ
```

revealing the definition of `×` used in the goal, about which more shortly, but also telling us about the types and visibility of variables in the *context*.

Our next move is to *split* one of the inputs into cases. We can see from the type information `ss : Vec .S .n` that we do not know the length of `ss`, so it might be given by either constructor. To see if Agda agrees, we type `ss` in the hole and issue the ‘case-split’ command `[C - c C - c]`.

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ss ts = { ss [C - c C - c] }0
```

Agda responds by editing our source code, replacing the single line of definition by two more specific cases.

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ⟨⟩ ts = { }0
zip (x, ss) ts = { }1
```

Moreover, we gain the refined type information

```
?0 : Vec (.S × .T) 0
?1 : Vec (.S × .T) (suc .n)
```

which goes to show that the type system is now tracking what information is learned about the problem by inspecting *ss*. This capacity for *learning by testing* is the paradigmatic characteristic of dependently typed programming.

Now, when we split *ts* in the *0* case, we get

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ⟨⟩ ⟨⟩ = { }0
zip (x, ss) ts = { }1
```

and in the *suc* case,

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ⟨⟩ ⟨⟩ = { }0
zip (x, ss) (x1, ts) = { }1
```

It's not even as clever as Epigram.

as the more specific type now determines the shape. Sadly, Agda is not very clever about choosing names, but let us persevere. We have now made sufficient analysis of the input to determine the output, and shape-indexing has helpfully ruled out shape mismatch. It is now so obvious what must be output that Agda can figure it out for itself. If we issue the keystroke $[C - c C - a]$ in each hole, a type-directed program search robot called 'Agsy' tries to find an expression which will fit in the hole, assembling it from the available information without further case analysis. We obtain a complete program.

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ⟨⟩ ⟨⟩ = ⟨⟩
zip (x, ss) (x1, ts) = (x, x1), zip ss ts
```

I tend to α -convert and realign such programs manually, yielding

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ⟨⟩ ⟨⟩ = ⟨⟩
zip (s, ss) (t, ts) = (s, t), zip ss ts
```

What just happened? We made *Vec*, a version of *List*, indexed by *N*, and suddenly became able to work with 'elements in corresponding positions' with some degree of precision. That worked because *N* describes the *shape* of lists: indeed $N \cong \text{List One}$, instantiating the *List* element type to the type *One* with the single element *⟨⟩*, so that the only information present is the shape. Once we fix the shape, we acquire a fixed notion of position.

Exercise 1.1 (vec) Complete the implementation of

```
vec : forall {n X} → X → Vec X n
vec {n} x = ?
```

Why is there no specification?

using only control codes and arrow keys. (Note the brace notation, making the implicit *n* explicit. It is not unusual for arguments to be inferrable at usage sites from type information, but none the less computationally relevant.)

Exercise 1.2 (vector application) Complete the implementation of

```
vapp : forall {n S T} → Vec (S → T) n → Vec S n → Vec T n
vapp fs ss = ?
```

using only control codes and arrow keys. The function should apply the functions from its first input vector to the arguments in corresponding positions from its second input vector, yielding values in corresponding positions in the output.

Exercise 1.3 (vmap) Using `vec` and `vapp`, define the functorial ‘map’ operator for vectors, applying the given function to each element.

```
vmap : forall {n S T} → (S → T) → Vec S n → Vec T n
vmap f ss = ?
```

Note that you can make Agsy notice a defined function by writing its name as a hint in the relevant hole before you `[C - c C - a]`.

Exercise 1.4 (zip) Using `vec` and `vapp`, give an alternative definition of `zip`.

```
zip : forall {n S T} → Vec S n → Vec T n → Vec (S × T) n
zip ss ts = ?
```

Exercise 1.5 (Finite sets and projection from vectors) We may define a type of finite sets, suitable for indexing into vectors, as follows:

```
data Fin : ℕ → Set where
  zero : {n : ℕ} → Fin (suc n)
  suc : {n : ℕ} → Fin n → Fin (suc n)
```

Implement projection:

```
proj : forall {n X} → Vec X n → Fin n → X
proj xs i = ?
```

Implement, tabulation, the inverse of projection.

```
tabulate : forall {n X} → (Fin n → X) → Vec X n
tabulate {n} f = ?
```

Hint: think higher order.

1.3 Applicative and Traversable Structure

The `vec` and `vapp` operations from the previous section equip vectors with the structure of an *applicative functor*. Before we get to **Applicative**, let us first say what is an **EndoFunctor**:

```
record EndoFunctor (F : Set → Set) : Set₁ where
  field
    map : forall {S T} → (S → T) → F S → F T
  open EndoFunctor {F ...} public
```

The above record declaration creates new types **EndoFunctor** *F* and a new *module*, **EndoFunctor**, containing a function, **EndoFunctor.map**, which projects the `map` field from a record. The **open** declaration brings `map` into top level scope, and the `{{...}}` syntax indicates that `map`’s record argument is an *instance argument*. Instance arguments are found by searching the context for something of the required type, succeeding if exactly one candidate is found.

Of course, we should ensure that such structures should obey the functor laws, with `map` preserving identity and composition. Dependent types allow us to state and prove these laws, as we shall see shortly.

First, however, let us refine **EndoFunctor** to **Applicative**.

For now, I shall just work in **Set**, but we should remember to break out and live, categorically, later. Why **Set**₁?

```

record Applicative (F : Set → Set) : Set1 where
  infixl 2 ⊗_
  field
    pure : forall {X} → X → F X
    ⊗_ : forall {S T} → F (S → T) → F S → F T
    applicativeEndoFunctor : EndoFunctor F
    applicativeEndoFunctor = record { map = ⊗_ ∘ pure }
open Applicative { { ... } } public

```

The `Applicative F` structure decomposes F 's `map` as the ability to make 'constant' F -structures and closure under application.

Given that instance arguments are collected from the context, let us seed the context with suitable candidates for `Vec`:

```

applicativeVec : forall {n} → Applicative λ X → Vec X n
applicativeVec = record { pure = vec; ⊗_ = vapp }
endoFunctorVec : forall {n} → EndoFunctor λ X → Vec X n
endoFunctorVec = applicativeEndoFunctor

```

Indeed, the definition of `endoFunctorVec` already makes use of way *its* `EndoFunctor` searches the context and finds `applicativeVec`.

`proj` and `tabulate`
turn the `vec` and
`vapp` applicative
into this one.

There are lots of applicative functors about the place. Here's another famous one:

```

applicativeFun : forall {S} → Applicative λ X → S → X
applicativeFun = record
  { pure = λ x s → x          -- also known as K (drop environment)
  ; ⊗_ = λ f a s → f s (a s) -- also known as S (share environment)
  }

```

Monadic structure induces applicative structure:

```

record Monad (F : Set → Set) : Set1 where
  field
    return : forall {X} → X → F X
    >>= : forall {S T} → F S → (S → F T) → F T
    monadApplicative : Applicative F
    monadApplicative = record
      { pure = return
      ; ⊗_ = λ ff fs → ff >>= λ f → fs >>= λ s → return (f s) }
open Monad { { ... } } public

```

Exercise 1.6 (Vec monad) Construct a `Monad` satisfying the `Monad` laws

```

monadVec : { n : ℕ } → Monad λ X → Vec X n
monadVec = ?

```

such that `monadApplicative` agrees extensionally with `applicativeVec`.

Exercise 1.7 (Applicative identity and composition) Show by construction that the identity endofunctor is `Applicative`, and that the composition of `Applicatives` is `Applicative`.

```

applicativeld : Applicative id
applicativeld = ?
applicativeComp : forall {F G} → Applicative F → Applicative G → Applicative (F ∘ G)
applicativeComp aF aG = ?

```


Exercise 1.8 (Monoid makes Applicative) Let us give the signature for a monoid thus:

```
record Monoid (X : Set) : Set where
  infixr 4 _•_
  field
    ε      : X
    _•_    : X → X → X
    monoidApplicative : Applicative λ _ → X
    monoidApplicative = ?
  open Monoid { {...} } public -- it's not obvious that we'll avoid ambiguity
```

Complete the `Applicative` so that it behaves like the `Monoid`.

Exercise 1.9 (Applicative product) Show by construction that the pointwise product of `Applicatives` is `Applicative`.

```
record Traversable (F : Set → Set) : Set1 where
  field
    traverse : forall {G S T} { {AG : Applicative G} } →
      (S → G T) → F S → G (F T)
    traversableEndoFunctor : EndoFunctor F
    traversableEndoFunctor = record { map = traverse }
  open Traversable { {...} } public
```

```
traversableVec : {n : ℕ} → Traversable λ X → Vec X n
traversableVec = record { traverse = vtr } where
  vtr : forall {n G S T} { {_ : Applicative G} } →
    (S → G T) → Vec S n → G (Vec T n)
  vtr { {aG} } f ⟨ ⟩ = pure { {aG} } ⟨ ⟩
  vtr { {aG} } f (s, ss) = pure { {aG} } -, - ⊗ f s ⊗ vtr f ss
```

The explicit `aG` became needed after I introduced the `applicativeld` exercise, making resolution ambiguous.

Exercise 1.10 (transpose) Implement matrix transposition in one line.

```
transpose : forall {m n X} → Vec (Vec X n) m → Vec (Vec X m) n
transpose = ?
```

We may define the `crush` operation, accumulating values in a monoid stored in a `Traversable` structure:

```
crush : forall {F X Y} { {TF : Traversable F} } { {M : Monoid Y} } →
  (X → Y) → F X → Y
crush { {M = M} } =
  traverse { T = One } { {AG = monoidApplicative { {M} } } } -- T arbitrary arguments.
```

I was going to set this as an exercise, but it's mostly instructive in how to override implicit and instance arguments.

Amusingly, we must tell Agda which `T` is intended when viewing `X → Y` as `X → (λ _ → Y) T`. In a Hindley-Milner language, such uninferred things are unimportant because they are in any case parametric. In the dependently typed setting, we cannot rely on quantification being parametric (although in the absence of typecase, quantification over types cannot help so being).

Exercise 1.11 (Traversable functors) Show that `Traversable` is closed under identity and composition. What other structure does it preserve?

1.4 Σ -types and Other Equipment

Before we go any further, let us establish that the type $\Sigma (S : \text{Set}) (T : S \rightarrow \text{Set})$ has elements $(s : S), (t : T s)$, so that the type of the second component depends on the value of the first. From $p : \Sigma S T$, we may project $\text{fst } p : S$ and $\text{snd } p : T (\text{fst } p)$, but I also define \vee to be a low precedence uncurrying operator, so that $\vee \lambda s t \rightarrow \dots$ gives access to the components.

On the one hand, we may take $S \times T = \Sigma S \lambda _ \rightarrow T$ and generalize the binary product to its dependent version. On the other hand, we can see $\Sigma S T$ as generalising the binary sum to an S -ary sum, which is why the type is called Σ in the first place.

We can recover the binary sum (coproduct) by defining a two element type:

```
data Two : Set where tt ff : Two
```

It is useful to define a conditional operator, indulging my penchant for giving infix operators three arguments,

```
<?>- : forall {l} {P : Two → Set l} → P tt → P ff → (b : Two) → P b
(t <?> f) tt = t
(t <?> f) ff = f
```

for we may then define:

```
⊕- : Set → Set → Set
S + T = Σ Two (S <?> T)
```

Note that $\langle ? \rangle$ has been defined to work at all levels of the predicative hierarchy, so that we can use it to choose between **Sets**, as well as between ordinary values. Σ thus models both choice and pairing in data structures. That is, Σ generalizes binary product to the dependent case, and binary sum to arbitrary arity. I advise calling a Σ -type neither a ‘dependent sum’ nor a ‘dependent product’ (for a dependent function type is a something-adic product), but rather a ‘dependent pair type’.

1.5 Arithmetic

I don’t know about you, but I find I do a lot more arithmetic with types than I do with numbers, which is why I have used \times and $+$ for **Sets**. However, we shall soon need a little arithmetic for the sizes of things.

Exercise 1.12 (unary arithmetic) *Implement addition and multiplication for numbers.*

```
⊕N- : ℕ → ℕ → ℕ
x ⊕N y = ?
×N- : ℕ → ℕ → ℕ
x ×N y = ?
```

1.6 Normal Functors

A *normal* functor is given, up to isomorphism, by a set of *shapes* and a function which assigns to each shape a *size*. It is interpreted as the *dependent pair* of a shape, s , and a vector of elements whose length is the size of s .

```

record Normal : Set1 where
  constructor /-
  field
    Shape : Set
    size   : Shape → ℕ
    [ ]N : Set → Set
    [ ]N X = Σ Shape λ s → Vec X (size s)
open Normal public
infixr 0 /-

```

Let us have two examples. Vectors are the normal functors with a unique shape. Lists are the normal functors whose shape is their size.

```

VecN : ℕ → Normal
VecN n = One / pure n
ListN : Normal
ListN = ℕ / id

```

But let us not get ahead of ourselves. We can build a kit for normal functors corresponding to the type constructors that we often define, then build up composite structures. For example, let us have that constants and the identity are **Normal**.

```

KN : Set → Normal
KN A = A / λ _ → 0
IKN : Normal
IKN = VecN 1

```

Let us construct sums and products of normal functors.

```

+_N : Normal → Normal → Normal
(ShF / szF) +_N (ShG / szG) = (ShF + ShG) / v szF ⟨?⟩ szG
×_N : Normal → Normal → Normal
(ShF / szF) ×_N (ShG / szG) = (ShF × ShG) / v λ f g → szF f +_N szG g

```

Of course, it is one thing to construct these binary operators on **Normal**, but quite another to show they are worthy of their names.

```

nlnj : forall {X} (F G : Normal) → [ F ]N X + [ G ]N X → [ F +_N G ]N X
nlnj F G (tt, ShF, xs) = (tt, ShF), xs
nlnj F G (ff, ShG, xs) = (ff, ShG), xs

```

Now, we could implement the other direction of the isomorphism, but an alternative is to define the *inverse image*.

```

data _-1 : {S T : Set} (f : S → T) : T → Set where
  from : (s : S) → f-1 f s

```

Let us now show that **nlnj** is surjective.

```

nCase : forall {X} F G (s : [ F +_N G ]N X) → nlnj F G-1 s
nCase F G ((tt, ShF), xs) = from (tt, ShF), xs
nCase F G ((ff, ShG), xs) = from (ff, ShG), xs

```

That is, we have written more or less the other direction of the iso, but we have acquired some of the correctness proof for the cost of asking. We shall check that **nlnj** is injective shortly, once we have suitable equipment to say so.

The inverse of ‘nInj’ can be computed by nCase thus:

```
nOut : forall {X} (F G : Normal) → [ F +N G ]N X → [ F ]N X + [ G ]N X
nOut F G xs' with nCase F G xs'
nOut F G . (nInj F G xs) | from xs = xs
```

The **with** notation allows us to compute some useful information and add it to the collection of things available for inspection in pattern matching. By matching the result of nCase $F\ G\ xs'$ as **from** xs , we discover that *ipso facto*, xs' is nInj xs . It is in the nature of dependent types that inspecting one piece of data can refine our knowledge of the whole programming problem, hence McKinna and I designed **with** as a syntax for bringing new information to the problem. The usual Burstallian ‘case expression’ focuses on one scrutinee and shows us its refinements, but hides from us the refinement of the rest of the problem: in simply typed programming there is no such refinement, but here there is. Agda prefixes with a dot those parts of patterns, not necessarily linear constructor forms, which need not be checked dynamically because the corresponding value must be as indicated in any well typed usage.

Exercise 1.13 (normal pairing) Implement the constructor for normal functor pairs. It may help to define vector concatenation.

```
-++- : forall {m n X} → Vec X m → Vec X n → Vec X (m +N n)
xs ++ ys = ?
nPair : forall {X} (F G : Normal) → [ F ]N X × [ G ]N X → [ F ×N G ]N X
nPair F G fxx = ?
```

Show that your constructor is surjective.

Exercise 1.14 (ListN monoid) While you are in this general area, construct (from readily available components) the usual monoid structure for our normal presentation of lists.

```
listNMonoid : {X : Set} → Monoid ([ ListN ]N X)
listNMonoid = ?
```

We have already seen that the identity functor $\text{VecN } 1$ is **Normal**, but can we define composition?

```
∘N- : Normal → Normal → Normal
F ∘N (ShG / szG) = ? / ?
```

To choose the shape for the composite, we need to know the outer shape, and then the inner shape at each element position. That is:

```
∘N- : Normal → Normal → Normal
F ∘N (ShG / szG) = [ F ]N ShG / {!!}
```

Now, the composite must have a place for each element of each inner structure, so the size of the whole is the sum of the sizes of its parts. That is to say, we must traverse the shape, summing the sizes of each inner shape therein. Indeed, we can use **traverse**, given that \mathbb{N} is a monoid for $+_{\mathbb{N}}$ and that **Normal** functors are traversable because vectors are.

```
sumMonoid : Monoid N
sumMonoid = record {ε = 0; -•- = +N}
```

```

normalTraversable : (F : Normal) → Traversable [ F ]N
normalTraversable F = record
  { traverse = λ { { aG } } f → ∇ λ s xs → pure { { aG } } (−, − s) ⊗ traverse f xs }

```

Armed with this structure, we can implement the composite size operator as a `crush`.

```

⊞N : Normal → Normal → Normal
F ⊞N (ShG / szG) = [ F ]N ShG / crush { { normalTraversable F } } szG

```

The fact that we needed only the `Traversable` interface to F is a bit of a clue to a connection between `Traversable` and `Normal` functors. `Traversable` structures have a notion of size induced by the `Monoid` structure for \mathbb{N} :

```

sizeT : forall { F } { { TF : Traversable F } } { X } → F X → ℕ
sizeT = crush (λ _ → 1)

```

Hence, every `Traversable` functor has a `Normal` counterpart

```

normalT : forall F { { TF : Traversable F } } → Normal
normalT F = F One / sizeT

```

where the shape is an F with placeholder elements and the size is the number of such places.

Can we put a `Traversable` structure into its `Normal` representation? We can certainly extract the shape:

```

shapeT : forall { F } { { TF : Traversable F } } { X } → F X → F One
shapeT = traverse (λ _ → ⟨⟩)

```

We can also define the list of elements, which should have the same length as the size

```

one : forall { X } → X → [ ListN ]N X
one x = 1, (x, ⟨⟩)
contentsT : forall { F } { { TF : Traversable F } } { X } → F X → [ ListN ]N X
contentsT = crush one

```

and then try

```

toNormal : forall { F } { { TF : Traversable F } } { X } → F X → [ normalT F ]N X
toNormal fx = BAD (shapeT fx, snd (contentsT fx))

```

but it fails to typecheck because the size of the shape of fx is not obviously the length of the contents of fx . The trouble is that `Traversable F` is underspecified. In due course, we shall discover that it means just that F is naturally isomorphic to $[\text{normalT } F]_{\mathbb{N}}$. To see this, however, we shall need the capacity to reason equationally, which must wait until the next section. Check this.

Exercise 1.15 (normal morphisms) A normal morphism is given as follows

```

→N : Normal → Normal → Set
F →N G = (s : Shape F) → [ G ]N (Fin (size F s))

```

where any such thing determines a natural transformation from F to G .

```

nMorph : forall { F G } → F →N G → forall { X } → [ F ]N X → [ G ]N X
nMorph f (s, xs) with f s
...      | s', is = s', map (proj xs) is

```

Show how to compute the normal morphism representing a given natural transformation.

$\text{morphN} : \text{forall } \{F\ G\} \rightarrow (\text{forall } \{X\} \rightarrow \llbracket F \rrbracket_N X \rightarrow \llbracket G \rrbracket_N X) \rightarrow F \rightarrow_N G$
 $\text{morphN } f\ s = ?$

Exercise 1.16 (Hancock’s tensor) *Let*

$\otimes : \text{Normal} \rightarrow \text{Normal} \rightarrow \text{Normal}$
 $(\text{Sh}F / \text{sz}F) \otimes (\text{Sh}G / \text{sz}G) = (\text{Sh}F \times \text{Sh}G) / \forall \lambda f\ g \rightarrow \text{sz}F\ f \times_N \text{sz}G\ g$

Construct normal morphisms:

$\text{swap} : (F\ G : \text{Normal}) \rightarrow (F \otimes G) \rightarrow_N (G \otimes F)$
 $\text{swap } F\ G\ x = ?$
 $\text{drop} : (F\ G : \text{Normal}) \rightarrow (F \otimes G) \rightarrow_N (F \otimes_N G)$
 $\text{drop } F\ G\ x = ?$

*Hint: for **swap**, you may find you need to build some operations manipulating matrices.*
*Hint: for **drop**, it may help to prove a theorem about multiplication (see next section for details of equality), but you can get away without so doing.*

1.7 Proving Equations

Never trust a type theorist who has not changed their mind about equality at least once.

The best way to start a fight in a room full of type theorists is to bring up the topic of *equality*. There’s a huge design space, not least because we often have *two* notions of equality to work with, so we need to design both and their interaction.

On the one hand, we have *judgmental* equality. Suppose you have $s : S$ and you want to put s where a value of type T is expected. Can you? You can if $S \equiv T$. Different systems specify \equiv differently. Before dependent types arrived, syntactic equality (perhaps up to α -conversion) was often enough.

In dependently typed languages, it is quite convenient if $\text{Vec } X\ (2 + 2)$ is the same type as $\text{Vec } X\ 4$, so we often consider types up to the $\alpha\beta$ -conversion of the λ -calculus further extended by the defining equations of total functions. If we’ve been careful enough to keep the *open-terms* reduction of the language strongly normalizing, then \equiv is decidable, by normalize-and-compare in theory and by more carefully tuned heuristics in practice.

Agda takes things a little further by supporting η -conversion at some ‘negative’ types—specifically, function types and record types—where a type-directed and terminating η -expansion makes sense. Note that a *syntax-directed* ‘tit-for-tat’ approach, e.g. testing $f \equiv \lambda x \rightarrow t$ by testing $x \vdash f\ x \equiv t$ or $p \equiv (s, t)$ by $\text{fst } p \equiv s$ and $\text{snd } p \equiv t$, works fine because two non-canonical functions and pairs are equal if and only if their expansions are. But if you want the *eta*-rule for **One**, you need a cue to notice that $u \equiv v$ when both inhabit **One** and neither is $\langle \rangle$.

It is always tempting (hence, dangerous) to try to extract more work from the computer by making judgmental equality admit more equations which we consider morally true, but it is clear that any *decidable* judgmental equality will always disappoint—extensional equality of functions is undecidable, for example. Correspondingly, the equational theory of *open* terms (conceived as functions from valuations of their variables) will always be to some extent beyond the ken of the computer.

The remedy for our inevitable disappointment with judgmental equality is to define a notion of *evidence* for equality. It is standard practice to establish decidable certificate-checking for undecidable problems, and we have a standard mechanism for so doing—checking types. Let us have types $s \simeq t$ inhabited by proofs

that s and t are equal. We should ensure that $t \simeq t$ for all t , and that for all P , $s \simeq t \rightarrow P s \rightarrow P t$, in accordance with the philosophy of Leibniz. On this much, we may agree. But after that, the fight starts.

The above story is largely by way of an apology for the following declaration.

```
data  $\simeq$  {  $l$  } {  $X$  : Set  $l$  } (  $x$  :  $X$  ) :  $X \rightarrow$  Set  $l$  where
  refl :  $x \simeq x$ 
infix 1  $\simeq$ 
```

The size of equality types is also moot. Agda would allow us to put $s \simeq t$ in Set, however large s and t may be...

We may certainly implement Leibniz's rule.

```
subst : forall {  $k$   $l$  } {  $X$  : Set  $k$  } {  $s$   $t$  :  $X$  }  $\rightarrow$ 
   $s \simeq t \rightarrow (P : X \rightarrow$  Set  $l) \rightarrow P s \rightarrow P t$ 
subst refl  $P p = p$ 
```

The only canonical proof of $s \simeq t$ is **refl**, available only if $s \equiv t$, so we have declared that the equality predicate for *closed* terms is whatever judgmental equality we happen to have chosen. We have sealed our disappointment in, but we have gained the ability to prove useful equations on *open* terms. Moreover, the restriction to the judgmental equality is fundamental to the computational behaviour of our **subst** implementation: we take $p : P s$ and we return it unaltered as $p : P t$, so we need to ensure that $P s \equiv P t$, and hence that $s \equiv t$. If we want to make \simeq larger than \equiv , we need a more invasive approach to transporting data between provably equal types. For now, let us acknowledge the problem and make do.

We may register equality with Agda, via the following pragmas,

```
{-# BUILTIN EQUALITY _==_ #-}
{-# BUILTIN REFL refl #-}
```

...but for this pragma, we need $\simeq \{l\} \{X\} s t : \text{Set } l$

and thus gain access to Agda's support for equational reasoning.

Now that we have some sort of equality, we can specify laws for our structures, e.g., for **Monoid**.

```
record MonoidOK  $X$  { {  $M$  : Monoid  $X$  } } : Set where
  field
    absorbL : (  $x$  :  $X$  )  $\rightarrow$   $\varepsilon \bullet x \simeq x$ 
    absorbR : (  $x$  :  $X$  )  $\rightarrow$   $x \bullet \varepsilon \simeq x$ 
    assoc : (  $x$   $y$   $z$  :  $X$  )  $\rightarrow$  (  $x \bullet y$  )  $\bullet z \simeq x \bullet ( y \bullet z )$ 
```

Let's check that $+\mathbb{N}$ really gives a monoid.

```
natMonoidOK : MonoidOK  $\mathbb{N}$ 
natMonoidOK = record
  { absorbL =  $\lambda \_ \rightarrow$  refl
  ; absorbR =  $\_ +$  zero
  ; assoc = assoc+
  } where -- see below
```

The **absorbL** law follows by computation, but the other two require inductive proof.

```
 $\_ +$ zero : forall  $x \rightarrow x +_{\mathbb{N}} \text{zero} \simeq x$ 
zero  $\_ +$ zero = refl
suc  $n +$ zero rewrite  $n +$ zero = refl

assoc+ : forall  $x$   $y$   $z \rightarrow (x +_{\mathbb{N}} y) +_{\mathbb{N}} z \simeq x +_{\mathbb{N}} (y +_{\mathbb{N}} z)$ 
assoc+ zero  $y$   $z$  = refl
assoc+ (suc  $x$ )  $y$   $z$  rewrite assoc+  $x$   $y$   $z = refl$ 
```

The usual inductive proofs become structurally recursive functions, pattern matching on the argument in which $+\mathbb{N}$ is strict, so that computation unfolds. Sadly, an Agda program, seen as a proof document does not show you the subgoal structure. However, we can see that the base case holds computationally and the step case becomes trivial once we have rewritten the goal by the inductive hypothesis (being the type of the structurally recursive call).

differently from the way in which a Coq script also does not

Exercise 1.17 (ListN monoid) *This is a nasty little exercise. By all means warm up by proving that List X is a monoid with respect to concatenation, but I want you to have a crack at*

```
listNMonoidOK : { X : Set } → MonoidOK ([ ListN ]N X)
listNMonoidOK { X } = ?
```

*Hint 1: use curried helper functions to ensure structural recursion. The inductive step cases are tricky because the hypotheses equate number-vector pairs, but the components of those pairs are scattered in the goal, so **rewrite** will not help. Hint 2: use **subst** with a predicate of form $\forall \lambda n\ xs \rightarrow \dots$, which will allow you to abstract over separated places with n and xs .*

Exercise 1.18 (a not inconsiderable problem) *Find out what goes wrong when you try to state associativity of vector $++$, let alone prove it. What does it tell you about our \simeq setup?*

A monoid homomorphism is a map between their carrier sets which respects the operations.

```
record MonoidHom { X } { { MX : Monoid X } } { Y } { { MY : Monoid Y } } (f : X → Y) : Set where
  field
    respε : f ε ≃ ε
    resp• : forall x x' → f (x • x') ≃ f x • f x'
```

For example, taking the length of a list is, in the **Normal** representation, trivially a homomorphism.

```
fstHom : forall { X } → MonoidHom ([ ListN ]N X) { N } fst
fstHom = record { respε = refl; resp• = λ _ _ → refl }
```

Moving along to functorial structures, let us explore laws about the transformation of *functions*. Equations at higher order mean trouble ahead!

```
record EndoFunctorOK F { { FF : EndoFunctor F } } : Set1 where
  field
    endoFunctorId : forall { X } →
      map { { FF } } { X } id ≃ id
    endoFunctorCo : forall { R S T } (f : S → T) (g : R → S) →
      map { { FF } } f ∘ map g ≃ map (f ∘ g)
```

However, when we try to show,

```
vecEndoFunctorOK : forall { n } → EndoFunctorOK λ X → Vec X n
vecEndoFunctorOK = record
  { endoFunctorId = { }0
  ; endoFunctorCo = λ f g → { }1
  }
```

we see concrete goals (up to some tidying):


```

?0 : vapp (vec id) ≈ id
?1 : vapp (vec f) ∘ vapp (vec g) ≈ vapp (vec (f ∘ g))

```

This is a fool's errand. The pattern matching definition of `vapp` will not allow these equations on functions to hold at the level of \equiv . We could make them a little more concrete by doing induction on n , but we will still not force enough computation. Our \approx cannot be extensional for functions because it has canonical proofs for nothing more than \equiv , and \equiv cannot incorporate extensionality and remain decidable.

Some see this as reason enough to abandon decidability of \equiv , thence of type-checking.

We can define *pointwise* equality,

```

≐ : forall {l} {S : Set l} {T : S → Set l}
    (f g : (x : S) → T x) → Set l
f ≐ g = forall x → f x ≈ g x
infix 1 ≐

```

which is reflexive but not substitutive.

Now we can at least require:

```

record EndoFunctorOKP F { {FF : EndoFunctor F} } : Set1 where
  field
    endoFunctorId : forall {X} →
      map { {FF} } {X} id ≐ id
    endoFunctorCo : forall {R S T} (f : S → T) (g : R → S) →
      map { {FF} } f ∘ map g ≐ map (f ∘ g)

```

Exercise 1.19 (Vec functor laws) *Show that vectors are functorial.*

```

vecEndoFunctorOKP : forall {n} → EndoFunctorOKP λ X → Vec X n
vecEndoFunctorOKP = ?

```

1.8 Laws for Applicative and Traversable

Developing the laws for `Applicative` and `Traversable` requires more substantial chains of equational reasoning. Here are some operators which serve that purpose, inspired by work from Lennart Augustsson and Shin-Cheng Mu.

```

-=[_] : forall {l} {X : Set l} (x : X) {y z} → x ≈ y → y ≈ z → x ≈ z
-=[ refl ] q = q
-[_]_ : forall {l} {X : Set l} (x : X) {y z} → y ≈ x → y ≈ z → x ≈ z
- [ refl ]_ q = q
_□ : forall {l} {X : Set l} (x : X) → x ≈ x
x □ = refl
infixr 1 -=[_] -[_]_ □

```

These three build right-nested chains of equations. Each requires an explicit statement of where to start. The first two step along an equation used left-to-right or right-to-left, respectively, then continue the chain. Then, $x \square$ marks the end of the chain.

Meanwhile, we may need to rewrite in a context whilst building these proofs. In the expression syntax, we have nothing like `rewrite`.

```

cong : forall {k l} {X : Set k} {Y : Set l} (f : X → Y) {x y} → x ≈ y → f x ≈ f y
cong f refl = refl

```

Thus armed, let us specify what makes an **Applicative** acceptable, then show that such a thing is certainly a *Functor*.

I had to η -expand \circ in lieu of subtyping.

```
record ApplicativeOKP F { { AF : Applicative F } } : Set1 where
  field
    lawId : forall { X } (x : F X) →
      pure { { AF } } id ⊗ x ≈ x
    lawCo : forall { R S T } (f : F (S → T)) (g : F (R → S)) (r : F R) →
      pure { { AF } } (λ f g → f ∘ g) ⊗ f ⊗ g ⊗ r ≈ f ⊗ (g ⊗ r)
    lawHom : forall { S T } (f : S → T) (s : S) →
      pure { { AF } } f ⊗ pure s ≈ pure (f s)
    lawCom : forall { S T } (f : F (S → T)) (s : S) →
      f ⊗ pure s ≈ pure { { AF } } (λ f → f s) ⊗ f
  applicativeEndoFunctorOKP : EndoFunctorOKP F { { applicativeEndoFunctor } }
  applicativeEndoFunctorOKP = record
    { endoFunctorId = lawId
    ; endoFunctorCo = λ f g r →
      pure { { AF } } f ⊗ (pure { { AF } } g ⊗ r)
      < lawCo (pure f) (pure g) r >=
      pure { { AF } } (λ f g → f ∘ g) ⊗ pure f ⊗ pure g ⊗ r
      ≡< cong (λ x → x ⊗ pure g ⊗ r) (lawHom (λ f g → f ∘ g) f) >
      pure { { AF } } (_ ∘ f) ⊗ pure g ⊗ r
      ≡< cong (λ x → x ⊗ r) (lawHom (_ ∘ f) g) >
      pure { { AF } } (f ∘ g) ⊗ r
      □
    }
}
```

Exercise 1.20 (ApplicativeOKP for Vec) Check that vectors are properly applicative. You can get away with **rewrite** for these proofs, but you might like to try the new tools.

```
vecApplicativeOKP : { n : ℕ } → ApplicativeOKP λ X → Vec X n
vecApplicativeOKP = ?
```

Given that **traverse** is parametric in an **Applicative**, we should expect to observe the corresponding naturality. We thus need a notion of *applicative homomorphism*, being a natural transformation which respects **pure** and \otimes . That is,

```
→_ : forall (F G : Set → Set) → Set1
F → G = forall { X } → F X → G X
record AppHom { F } { { AF : Applicative F } } { G } { { AG : Applicative G } }
  (k : F → G) : Set1 where
  field
    resppure : forall { X } (x : X) → k (pure x) ≈ pure x
    resp⊗ : forall { S T } (f : F (S → T)) (s : F S) → k (f ⊗ s) ≈ k f ⊗ k s
```

We may readily check that monoid homomorphisms lift to applicative homomorphisms.

```
monoidApplicativeHom :
  forall { X } { { MX : Monoid X } } { Y } { { MY : Monoid Y } }
  (f : X → Y) { { hf : MonoidHom f } } →
  AppHom { { monoidApplicative { { MX } } } } { { monoidApplicative { { MY } } } } f
  monoidApplicativeHom f { { hf } } = record
```

```

{ resppure = λ x → MonoidHom.respε hf
; resp⊗    = MonoidHom.resp • hf
}

```

Exercise 1.21 (homomorphism begets applicative) Show that a homomorphism from F to G induces applicative structure on their pointwise sum.

```

homSum : forall {F G} { {AF : Applicative F} } { {AG : Applicative G} } →
  (f : F → G) →
  Applicative λ X → F X + G X
homSum { {AF} } { {AG} } f = ?

```

Check that your solution obeys the laws.

```

homSumOKP : forall {F G} { {AF : Applicative F} } { {AG : Applicative G} } →
  ApplicativeOKP F → ApplicativeOKP G →
  (f : F → G) → AppHom f →
  ApplicativeOKP _ { {homSum f} }
homSumOKP { {AF} } { {AG} } FOK GOK f homf = ?

```

Laws for `Traversable` functors are given thus:

```

record TraversableOKP F { {TF : Traversable F} } : Set1 where
  field
    lawId   : forall {X} (xs : F X) → traverse id xs ≈ xs
    lawCo   : forall {G} { {AG : Applicative G} } {H} { {AH : Applicative H} }
      { R S T } (g : S → G T) (h : R → H S) (rs : F R) →
      let EH : EndoFunctor H; EH = applicativeEndoFunctor
      in map {H} (traverse g) (traverse h rs)
      ≈
      traverse { {TF} } { {applicativeComp AH AG} } (map {H} g ∘ h) rs
    lawHom  : forall {G} { {AG : Applicative G} } {H} { {AH : Applicative H} }
      (h : G → H) { S T } (g : S → G T) → AppHom h →
      (ss : F S) →
      traverse (h ∘ g) ss ≈ h (traverse g ss)

```

Let us now check the coherence property we needed earlier.

```

lengthContentsSizeShape :
  forall {F} { {TF : Traversable F} } → TraversableOKP F →
  forall {X} (fx : F X) →
  fst (contentsT fx) ≈ sizeT (shapeT fx)
lengthContentsSizeShape tokF fx =
  fst (contentsT fx)
  < TraversableOKP.lawHom tokF { {monoidApplicative} } { {monoidApplicative} }
    fst one (monoidApplicativeHom fst) fx ≈
  sizeT fx
  < TraversableOKP.lawCo tokF { {monoidApplicative} } { {applicativeld} }
    (λ _ → 1) (λ _ → ⟨⟩) fx ≈
  sizeT (shapeT fx) □

```

We may now construct

```

toNormal : forall {F} { {TF : Traversable F} } → TraversableOKP F →
  forall {X} → F X → [normalT F]N X

```

```

toNormal tokf fx
= shapeT fx
, subst (lengthContentsSizeShape tokf fx) (Vec _) (snd (contentsT fx))

```

Exercise 1.22 Define `fromNormal`, reversing the direction of `toNormal`. One way to do it is to define what it means to be able to build something from a batch of contents.

```

Batch : Set → Set → Set
Batch X Y = Σ ℕ λ n → Vec X n → Y

```

Show `Batch X` is applicative. You can then use `traverse` on a `shape` to build a `Batch` job which reinserts the contents. As above, you will need to prove a coherence property to show that the contents vector in your hand has the required length. Warning: you may encounter a consequence of defining `sizeT` via `crush` with ignored target type `One`, and need to prove that you get the same answer if you ignore something else. Agda’s ‘Toggle display of hidden arguments’ menu option may help you detect that scenario.

Showing that `toNormal` and `fromNormal` are mutually inverse looks like a tall order, given that the programs have been glued together with coherence conditions. At time of writing, it remains undone. When I see a mess like that, I wonder whether replacing indexing by the measure of size might help.

1.9 Fixpoints of Normal Functors

The universal first order simple datatype is given by taking the least fixpoint of a normal functor.

```

data Tree (N : Normal) : Set where
  ⟨_⟩ : [ N ]_N (Tree N) → Tree N

```

We may, for example, define the natural numbers this way:

```

NatT : Normal
NatT = Two / 0 ⟨?⟩ 1
zeroT : Tree NatT
zeroT = ⟨ tt, ⟨ ⟩ ⟩
sucT : Tree NatT → Tree NatT
sucT n = ⟨ ff, n, ⟨ ⟩ ⟩

```

Of course, to prove these are the natural numbers, we need the eliminator as well as the constructors.

Exercise 1.23 Prove the principle of induction for these numbers.

```

NatInd : forall {l} (P : Tree NatT → Set l) →
  P zeroT →
  ((n : Tree NatT) → P n → P (sucT n)) →
  (n : Tree NatT) → P n
NatInd P z s n = ?

```

Indeed, there’s a generic induction principle for the whole lot of these types. First, we need predicate transformer to generate the induction hypothesis.

$\text{All} : \text{forall } \{l\ X\} (P : X \rightarrow \text{Set } l) \{n\} \rightarrow \text{Vec } X\ n \rightarrow \text{Set } l$
 $\text{All } P \langle \rangle = \text{One}$
 $\text{All } P (x, xs) = P\ x \times \text{All } P\ xs$

We then acquire

$\text{induction} : \text{forall } (N : \text{Normal}) \{l\} (P : \text{Tree } N \rightarrow \text{Set } l) \rightarrow$
 $((s : \text{Shape } N) (ts : \text{Vec } (\text{Tree } N) (\text{size } N\ s)) \rightarrow \text{All } P\ ts \rightarrow P\ \langle s, ts \rangle) \rightarrow$
 $(t : \text{Tree } N) \rightarrow P\ t$
 $\text{induction } N\ P\ p\ \langle s, ts \rangle = p\ s\ ts\ (\text{hyps } ts) \text{ where}$
 $\text{hyps} : \text{forall } \{n\} (ts : \text{Vec } (\text{Tree } N)\ n) \rightarrow \text{All } P\ ts$
 $\text{hyps } \langle \rangle = \langle \rangle$
 $\text{hyps } (t, ts) = \text{induction } N\ P\ p\ t, \text{hyps } ts$

Exercise 1.24 (decidable equality) We say a property is decided if we know whether it is true or false, where falsity is indicated by function to **Zero**, an empty type.

$\text{Dec} : \text{Set} \rightarrow \text{Set}$
 $\text{Dec } X = X + (X \rightarrow \text{Zero})$

Show that if a normal functor has decidable equality for its shapes, then its fixpoint also has decidable equality.

$\text{eq?} : (N : \text{Normal}) (\text{sheq?} : (s\ s' : \text{Shape } N) \rightarrow \text{Dec } (s \simeq s')) \rightarrow$
 $(t\ t' : \text{Tree } N) \rightarrow \text{Dec } (t \simeq t')$
 $\text{eq? } N\ \text{sheq? } t\ t' = ?$

Chapter 2

Simply Typed λ -Calculus

This chapter contains some standard techniques for the representation of typed syntax and its semantics. The joy of typed syntax is the avoidance of junk in its interpretation. Everything fits, just so.

2.1 Syntax

Last century, I learned the following recipe for well typed terms of the simply typed λ -calculus from Altenkirch and Reus.

First, give a syntax for types. I shall start with a base type and close under function spaces.

```
data ★ : Set where
  !   : ★
  ▷_  : ★ → ★ → ★
infixr 5 ▷_
```

Next, build contexts as snoc-lists.

```
data Cx (X : Set) : Set where
  ε   : Cx X
  _⊢ : Cx X → X → Cx X
infixl 4 ⊢
```

Now, define typed de Bruijn indices to be context membership evidence.

```
data ∈ (τ : ★) : Cx ★ → Set where
  zero : forall {Γ} → τ ∈ Γ, τ
  suc   : forall {Γ σ} → τ ∈ Γ → τ ∈ Γ, σ
infix 3 ∈
```

That done, we can build well typed terms by writing syntax-directed rules for the typing judgment.

```
data ⊢_ (Γ : Cx ★) : ★ → Set where
  var : forall {τ}
    → τ ∈ Γ
    -----
    → Γ ⊢ τ
  lam : forall {σ τ}
    → Γ ⊢ σ
    → Γ ⊢ τ
```

$$\begin{array}{c}
\rightarrow \Gamma, \sigma \vdash \tau \\
\hline
\rightarrow \Gamma \vdash \sigma \triangleright \tau \\
\text{app} : \text{forall } \{\sigma \tau\} \\
\rightarrow \Gamma \vdash \sigma \triangleright \tau \rightarrow \Gamma \vdash \sigma \\
\hline
\rightarrow \Gamma \vdash \tau \\
\text{infix } \beta \vdash_
\end{array}$$

2.2 Semantics

Writing an interpreter for such a calculus is an exercise also from last century, for which we should thank Augustsson and Carlsson. Start by defining the semantics of each type.

$$\begin{array}{l}
\llbracket \cdot \rrbracket_* : * \rightarrow \text{Set} \\
\llbracket \iota \rrbracket_* = \mathbb{N} \quad \text{-- by way of being nontrivial} \\
\llbracket \sigma \triangleright \tau \rrbracket_* = \llbracket \sigma \rrbracket_* \rightarrow \llbracket \tau \rrbracket_*
\end{array}$$

Next, define *environments* for contexts, with projection. We can reuse these definitions in the rest of the section if we abstract over the notion of value.

$$\begin{array}{l}
\llbracket \cdot \rrbracket_{Cx} : Cx * \rightarrow (* \rightarrow \text{Set}) \rightarrow \text{Set} \\
\llbracket \mathcal{E} \rrbracket_{Cx} V = \text{One} \\
\llbracket \Gamma, \sigma \rrbracket_{Cx} V = \llbracket \Gamma \rrbracket_{Cx} V \times V \sigma \\
\llbracket \cdot \rrbracket_{\in} : \text{forall } \{\Gamma \tau V\} \rightarrow \tau \in \Gamma \rightarrow \llbracket \Gamma \rrbracket_{Cx} V \rightarrow V \tau \\
\llbracket \text{zero} \rrbracket_{\in} (\gamma, t) = t \\
\llbracket \text{suc } i \rrbracket_{\in} (\gamma, s) = \llbracket i \rrbracket_{\in} \gamma
\end{array}$$

Finally, define the meaning of terms.

$$\begin{array}{l}
\llbracket \cdot \rrbracket_{\in} : \text{forall } \{\Gamma \tau\} \rightarrow \Gamma \vdash \tau \rightarrow \llbracket \Gamma \rrbracket_{Cx} \llbracket \cdot \rrbracket_* \rightarrow \llbracket \tau \rrbracket_* \\
\llbracket \text{var } i \rrbracket_{\vdash} \gamma = \llbracket i \rrbracket_{\in} \gamma \\
\llbracket \text{lam } t \rrbracket_{\vdash} \gamma = \lambda s \rightarrow \llbracket t \rrbracket_{\vdash} (\gamma, s) \\
\llbracket \text{app } f s \rrbracket_{\vdash} \gamma = \llbracket f \rrbracket_{\vdash} \gamma (\llbracket s \rrbracket_{\vdash} \gamma)
\end{array}$$

2.3 Substitution with a Friendly Fish

We may define the types of simultaneous renamings and substitutions as type-preserving maps from variables:

$$\begin{array}{l}
\text{Ren Sub} : Cx * \rightarrow Cx * \rightarrow \text{Set} \\
\text{Ren } \Gamma \Delta = \text{forall } \{\tau\} \rightarrow \tau \in \Gamma \rightarrow \tau \in \Delta \\
\text{Sub } \Gamma \Delta = \text{forall } \{\tau\} \rightarrow \tau \in \Gamma \rightarrow \Delta \vdash \tau
\end{array}$$

The trouble with defining the action of substitution for a de Bruijn representation is the need to shift indices when the context grows. Here is one way to address that situation. First, let me define context extension as concatenation with a cons-list, using the \hookrightarrow operator.

\hookrightarrow is pronounce ‘fish’, for historical reasons.

$$\begin{array}{l}
\hookrightarrow : \text{forall } \{X\} \rightarrow Cx X \rightarrow \text{List } X \rightarrow Cx X \\
xz \hookrightarrow \langle \rangle = xz
\end{array}$$

$xz \triangleleft (x, xs) = xz, x \triangleleft xs$
infixl 4 \triangleleft

We may then define the *shiftable* simultaneous substitutions from Γ to Δ as type-preserving mappings from the variables in any extension of Γ to terms in the same extension of Δ .

Shub : $Cx \star \rightarrow Cx \star \rightarrow \text{Set}$
Shub $\Gamma \Delta = \text{forall } \Xi \rightarrow \text{Sub } (\Gamma \triangleleft \Xi) (\Delta \triangleleft \Xi)$

By the computational behaviour of \triangleleft , a **Shub** $\Gamma \Delta$ can be used as a **Shub** $(\Gamma, \sigma) (\Delta, \sigma)$, so we can push substitutions under binders very easily.

$_//_ : \text{forall } \{ \Gamma \Delta \} (\theta : \text{Shub } \Gamma \Delta) \{ \tau \} \rightarrow \Gamma \vdash \tau \rightarrow \Delta \vdash \tau$
 $\theta // \text{var } i = \theta \langle \rangle i$
 $\theta // \text{lam } t = \text{lam } ((\theta \circ _, -) // t)$
 $\theta // \text{app } f s = \text{app } (\theta // f) (\theta // s)$

Of course, we shall need to construct some of these joyous shubstitutions. Let us first show that any simultaneous renaming can be made shiftable by iterative weakening.

wkr : $\text{forall } \{ \Gamma \Delta \sigma \} \rightarrow \text{Ren } \Gamma \Delta \rightarrow \text{Ren } (\Gamma, \sigma) (\Delta, \sigma)$
wkr $r \text{ zero} = \text{zero}$
wkr $r (\text{suc } i) = \text{suc } (r i)$
ren : $\text{forall } \{ \Gamma \Delta \} \rightarrow \text{Ren } \Gamma \Delta \rightarrow \text{Shub } \Gamma \Delta$
ren $r \langle \rangle = \text{var } \circ r$
ren $r (_, \Xi) = \text{ren } (\text{wkr } r) \Xi$

With renaming available, we can play the same game for substitutions.

wks : $\text{forall } \{ \Gamma \Delta \sigma \} \rightarrow \text{Sub } \Gamma \Delta \rightarrow \text{Sub } (\Gamma, \sigma) (\Delta, \sigma)$
wks $s \text{ zero} = \text{var zero}$
wks $s (\text{suc } i) = \text{ren suc } // s i$
sub : $\text{forall } \{ \Gamma \Delta \} \rightarrow \text{Sub } \Gamma \Delta \rightarrow \text{Shub } \Gamma \Delta$
sub $s \langle \rangle = s$
sub $s (_, \Xi) = \text{sub } (\text{wks } s) \Xi$

2.4 A Modern Convenience

Bob Atkey once remarked that ability to cope with de Bruijn indices was a good reverse Turing Test, suitable for detecting humaniform robotic infiltrators. Correspondingly, we might like to write terms which use real names. I had an idea about how to do that.

We can build the renaming which shifts past any context extension.

weak : $\text{forall } \{ \Gamma \} \Xi \rightarrow \text{Ren } \Gamma (\Gamma \triangleleft \Xi)$
weak $\langle \rangle i = i$
weak $(_, \Xi) i = \text{weak } \Xi (\text{suc } i)$

Then, we can observe that to build the body of a binder, it is enough to supply a function which will deliver the term representing the variable in any suitably extended context. The context extension is given implicitly, to be inferred from the usage site, and then the correct weakening is applied to the bound variable.

```

lambda : forall {Γ σ τ} →
  ((forall {Ξ} → Γ, σ << Ξ ⊢ σ) → Γ, σ ⊢ τ) →
  Γ ⊢ σ ▷ τ
lambda f = lam (f λ {Ξ} → var (weak Ξ zero))

```

But sadly, the following does not typecheck

```

myTest : E ⊢ ι ▷ ι
myTest = lambda λ x → x

```

because the following constraint is not solved:

$$(E, \iota << _Xi_232\ x) = (E, \iota) : Cx \star$$

That is, constructor-based unification is insufficient to solve for the prefix of a context, given a common suffix.

By contrast, solving for a suffix is easy when the prefix is just a value: it requires only the stripping off of matching constructors. So, we can cajole Agda into solving the problem by working with its reversal, via the ‘chips’ operator:

```

_<<_ : forall {X} → Cx X → List X → List X
E      << ys = ys
(xz, x) << ys = xz << (x, ys)

```

Of course, one must prove that solving the reverse problem is good for solving the original.

I have discovered a truly appalling proof of this lemma. Fortunately, this margin is too narrow to contain it. See if you can do better.

Exercise 2.1 (reversing lemma) *Show*

```

lem : forall {X} (Δ Γ : Cx X) Ξ →
  Δ << ⟨⟩ ≃ Γ << Ξ → Γ << Ξ ≃ Δ
lem Δ Γ Ξ q = ?

```

Now we can frame the constraint solve as an instance argument supplying a proof of the relevant equation on cons-lists: Agda will try to use `refl` to solve the instance argument, triggering the tractable version of the unification problem.

```

lambda : forall {Γ σ τ} →
  ((forall {Δ Ξ} { _ : Δ << ⟨⟩ ≃ Γ << (σ, Ξ) } → Δ ⊢ σ) →
  Γ, σ ⊢ τ) →
  Γ ⊢ σ ▷ τ
lambda {Γ} f =
  lam (f λ {Δ Ξ} { {q}} →
    subst (lem Δ Γ ( _ , Ξ ) q) (λ Γ → Γ ⊢ _ ) (var (weak Ξ zero)))
myTest : E ⊢ (ι ▷ ι) ▷ (ι ▷ ι)
myTest = lambda λ f → lambda λ x → app f (app f x)

```

2.5 Hereditary Substitution

This section is a structured series of exercises, delivering a $\beta\eta$ -long normalization algorithm for our λ -calculus by the method of *hereditary substitution*.

The target type for the algorithm is the following right-nested spine representation of β -normal η -long forms.

```

mutual
  data  $\vdash$  (Γ : Cx ★) : ★ → Set where
    lam : forall {σ τ} → Γ, σ ⊢ τ → Γ ⊢ σ ▷ τ
     $\underline{\$}$  : forall {τ} → τ ∈ Γ → Γ ⊢* τ → Γ ⊢  $\iota$ 

  data  $\vdash^*$  (Γ : Cx ★) : ★ → Set where
    ⟨ ⟩ : Γ ⊢*  $\iota$ 
     $\rightarrow$ ,  $\rightarrow$  : forall {σ τ} → Γ ⊢ σ → Γ ⊢* τ → Γ ⊢* σ ▷ τ

infix 3  $\vdash$   $\vdash^*$ 
infix 3  $\underline{\$}$ 

```

That is $\Gamma \vdash \tau$ is the type of normal forms in τ , and $\Gamma \vdash^* \tau$ is the type of spines for a τ , delivering ι .

The operation of hereditary substitution replaces *one* variable with a normal form and immediately performs all the resulting computation (i.e., more substitution), returning a normal form. You will need some equipment for talking about individual variables.

Exercise 2.2 (thinning) Define the function $-$ which removes a designated entry from a context, then implement the thinning operator, being the renaming which maps the embed the smaller context back into the larger.

```

 $\rightarrow$  : forall (Γ : Cx ★) {τ} (x : τ ∈ Γ) → Cx ★
Γ  $\rightarrow$  x = ?
infixl 4  $\rightarrow$ 

 $\neq$  : forall {Γ σ} (x : σ ∈ Γ) → Ren (Γ  $\rightarrow$  x) Γ
x  $\neq$  y = ?

```

This much will let us frame the problem. We have a candidate value for x which does not depend on x , so we should be able to eliminate x from any term by substituting out. If we try, we find this situation:

```

⟨  $\rightarrow$  ⟩ : forall {Γ σ τ} → (x : σ ∈ Γ) → Γ  $\rightarrow$  x ⊢ σ →
    Γ ⊢ τ → Γ  $\rightarrow$  x ⊢ τ
⟨ x ↦ s ⟩ lam t = lam (⟨ suc x ↦ ? ⟩ t)
⟨ x ↦ s ⟩ y $ ts = ?
infix 2 ⟨  $\rightarrow$  ⟩

```

Let us now address the challenges we face.

In the application case, we shall need to test whether or not y is the x for which we must substitute, so we need some sort of equality test. A *Boolean* equality test does not generate enough useful information—if y is x , we need to know that ts is a suitable spine for s ; if y is not x , we need to know its representation in $\Gamma \rightarrow x$. Hence, let us rather prove that any variable is either the one we are looking for or another. We may express this discriminability property as a predicate on variables.

```

data Veq? {Γ σ} (x : σ ∈ Γ) : forall {τ} → τ ∈ Γ → Set where
  same : Veq? x x
  diff : forall {τ} (y : τ ∈ Γ  $\rightarrow$  x) → Veq? x (x  $\neq$  y)

```

Exercise 2.3 (variable equality testing) Show that every y is discriminable with respect to a given x .

```

veq? : forall {Γ σ τ} (x : σ ∈ Γ) (y : τ ∈ Γ) → Veq? x y
veq? x y = ?

```

Hint: it will help to use **with** in the recursive case.

Meanwhile, in the **lam** case, we may easily shift x to account for the new variable in t , but we shall also need to shift s .

Exercise 2.4 (closure under renaming) Show how to propagate a renaming through a normal form.

mutual

```
renNm : forall {Γ Δ τ} → Ren Γ Δ → Γ ⊢ τ → Δ ⊢ τ
renNm r t = ?
renSp : forall {Γ Δ τ} → Ren Γ Δ → Γ ⊢* τ → Δ ⊢* τ
renSp r ss = ?
```

Now we have everything we need to implement hereditary substitution.

Exercise 2.5 (hereditary substitution) Implement hereditary substitution for normal forms and spines, defined mutually with application of a normal form to a spine, performing β -reduction.

mutual

```
<_→_>_ : forall {Γ σ τ} → (x : σ ∈ Γ) → Γ - x ⊢ σ →
      Γ ⊢ τ → Γ - x ⊢ τ
< x ↦ s > t = ?
<_→_>* _ : forall {Γ σ τ} → (x : σ ∈ Γ) → Γ - x ⊢ σ →
      Γ ⊢* τ → Γ - x ⊢* τ
< x ↦ s >* ts = ?
$ : forall {Γ τ} →
      Γ ⊢ τ → Γ ⊢* τ → Γ ⊢ ι
f $ ss = ?
infix 3 $
infix 2 <_→_>_
```

Do you think these functions are mutually structurally recursive?

With hereditary substitution, it should be a breeze to implement normalization, but there is one little tricky part remaining.

Exercise 2.6 (η -expansion for **normalize)** If we start implementing **normalize**, it is easy to get this far:

```
normalize : forall {Γ τ} → Γ ⊢ τ → Γ ⊢ τ
normalize (var x) = ?
normalize (lam t) = lam (normalize t)
normalize (app f s) with normalize f | normalize s
normalize (app f s) | lam t | s' = < zero ↦ s' > t
```

We can easily push under **lam** and implement **app** by hereditary substitution. However, if we encounter a variable, x , we must deliver it in η -long form. You will need to figure out how to expand x in a type-directed manner, which is not a trivial thing to do. Hint: if you need to represent the prefix of a spine, it suffices to consider functions from suffices.

Here are a couple of test examples for you to try. You may need to translate them into de Bruijn terms manually if you have not yet proven the ‘reversing lemma’.

```

try1 :  $\mathcal{E} \models ((\iota \triangleright \iota) \triangleright (\iota \triangleright \iota)) \triangleright (\iota \triangleright \iota) \triangleright (\iota \triangleright \iota)$ 
try1 = normalize (lambda  $\lambda x \rightarrow x$ )
church2 : forall { $\tau$ }  $\rightarrow \mathcal{E} \vdash (\tau \triangleright \tau) \triangleright \tau \triangleright \tau$ 
church2 = lambda  $\lambda f \rightarrow \text{lambda } \lambda x \rightarrow \text{app } f (\text{app } f x)$ 
try2 :  $\mathcal{E} \models (\iota \triangleright \iota) \triangleright (\iota \triangleright \iota)$ 
try2 = normalize (app (app church2 church2) church2)

```

2.6 Normalization by Evaluation

Let's cook normalization a different way, extracting more leverage from Agda's computation machinery. the idea is to model values as either 'going' (capable of computation if applied) or 'stopping' (incapable of computation, but not η -long). The latter terms look like left-nested applications of a variable.

```

data Stop ( $\Gamma : \text{Cx} \star$ ) ( $\tau : \star$ ) : Set where
  var :  $\tau \in \Gamma \rightarrow \text{Stop } \Gamma \tau$ 
   $\underline{s}$  : forall { $\sigma$ }  $\rightarrow \text{Stop } \Gamma (\sigma \triangleright \tau) \rightarrow \Gamma \models \sigma \rightarrow \text{Stop } \Gamma \tau$ 

```

Exercise 2.7 (Stop equipment) Show that `Stop` terms are closed under renaming, and that you can apply them to a spine to get a normal form.

```

renSt : forall { $\Gamma \Delta \tau$ }  $\rightarrow \text{Ren } \Gamma \Delta \rightarrow \text{Stop } \Gamma \tau \rightarrow \text{Stop } \Delta \tau$ 
renSt  $r u = ?$ 
stopSp : forall { $\Gamma \tau$ }  $\rightarrow \text{Stop } \Gamma \tau \rightarrow \Gamma \models^* \tau \rightarrow \Gamma \models \iota$ 
stopSp  $u ss = ?$ 

```

Let us now give a contextualized semantics to each type. Values either `Go` or `Stop`. Ground values cannot go: `Zero` is a datatype with no constructors. Functional values have a Kripke semantics. Wherever their context is meaningful, they take values to values.

```

mutual
  Val :  $\text{Cx} \star \rightarrow \star \rightarrow \text{Set}$ 
  Val  $\Gamma \tau = \text{Go } \Gamma \tau + \text{Stop } \Gamma \tau$ 
  Go :  $\text{Cx} \star \rightarrow \star \rightarrow \text{Set}$ 
  Go  $\Gamma \iota = \text{Zero}$ 
  Go  $\Gamma (\sigma \triangleright \tau) = \text{forall } \{ \Delta \} \rightarrow \text{Ren } \Gamma \Delta \rightarrow \text{Val } \Delta \sigma \rightarrow \text{Val } \Delta \tau$ 

```

Exercise 2.8 (renaming values and environments) Show that values admit renaming. Extend renaming to environments storing values. Construct the identity environment, mapping each variable to itself.

```

renVal : forall { $\Gamma \Delta$ }  $\tau \rightarrow \text{Ren } \Gamma \Delta \rightarrow \text{Val } \Gamma \tau \rightarrow \text{Val } \Delta \tau$ 
renVal  $\tau r v = ?$ 
renVals : forall  $\Theta$  { $\Gamma \Delta$ }  $\rightarrow \text{Ren } \Gamma \Delta \rightarrow \llbracket \Theta \rrbracket_{\text{Cx}} (\text{Val } \Gamma) \rightarrow \llbracket \Theta \rrbracket_{\text{Cx}} (\text{Val } \Delta)$ 
renVals  $\Theta r \theta = ?$ 
idEnv : forall  $\Gamma \rightarrow \llbracket \Gamma \rrbracket_{\text{Cx}} (\text{Val } \Gamma)$ 
idEnv  $\Gamma = ?$ 

```

Exercise 2.9 (application and quotation) Implement application for values. In order to apply a stopped function, you will need to be able to extract a normal form for the argument, so you will also need to be able to ‘quote’ values as normal forms.

It seems `quote` is a reserved symbol in Agda.

mutual

```

apply : forall {  $\Gamma$   $\sigma$   $\tau$  }  $\rightarrow$  Val  $\Gamma$  ( $\sigma \triangleright \tau$ )  $\rightarrow$  Val  $\Gamma$   $\sigma \rightarrow$  Val  $\Gamma$   $\tau$ 
apply f s = ?
quo : forall {  $\Gamma$  }  $\tau \rightarrow$  Val  $\Gamma$   $\tau \rightarrow \Gamma \vdash \tau$ 
quo  $\tau$  v = ?

```

For the last step, we need to compute values from terms.

Exercise 2.10 (evaluation) Show that every well typed term can be given a value in any context where its free variables have values.

```

eval : forall {  $\Gamma$   $\Delta$   $\tau$  }  $\rightarrow \Gamma \vdash \tau \rightarrow \llbracket \Gamma \rrbracket_{\text{Cx}} (\text{Val } \Delta) \rightarrow \text{Val } \Delta$   $\tau$ 
eval t  $\gamma$  = ?

```

With all the pieces in place, we get

```

normByEval : forall {  $\Gamma$   $\tau$  }  $\rightarrow \Gamma \vdash \tau \rightarrow \Gamma \vdash \tau$ 
normByEval {  $\Gamma$  } {  $\tau$  } t = quo  $\tau$  (eval t (idEnv  $\Gamma$ ))

```

Exercise 2.11 (numbers and primitive recursion) Consider extending the term language with constructors for numbers and a primitive recursion operator.

```

zero :  $\Gamma \vdash \mathbf{l}$ 
suc :  $\Gamma \vdash \mathbf{l} \rightarrow \Gamma \vdash \mathbf{l}$ 
rec : forall {  $\tau$  }  $\rightarrow \Gamma \vdash \tau \rightarrow \Gamma \vdash (\mathbf{l} \triangleright \tau \triangleright \tau)$ 
       $\rightarrow \Gamma \vdash \mathbf{l} \rightarrow \Gamma \vdash \tau$ 

```

How should the normal forms change? How should the values change? Can you extend the implementation of normalization?

Exercise 2.12 (adding adding) Consider making the further extension with a hardwired addition operator.

```

suc :  $\Gamma \vdash \mathbf{l} \rightarrow \Gamma \vdash \mathbf{l} \rightarrow \Gamma \vdash \mathbf{l}$ 

```

Can you engineer the notion of value and the evaluator so that `normByEval` identifies

add zero t	with	t
add s zero	with	s
add (suc s) t	with	suc (add s t)
add s (suc t)	with	suc (add s t)
add (add r s) t	with	add r (add s t)
add s t	with	add t s

and thus yields a stronger decision procedure for equality of expressions involving adding? (This is not an easy exercise, especially if you want the last equation to hold. I must confess I have not worked out the details.)

Chapter 3

Containers and W-types

Containers are the infinitary generalization of normal functors.

```

record Con : Set1 where
  constructor  $\triangleleft$ 
  field
    Sh : Set          -- a set of shapes
    Po : Sh → Set    -- a family of positions
     $\llbracket - \rrbracket_{\triangleleft}$  where : Set → Set
     $\llbracket - \rrbracket_{\triangleleft}$  where  $X = \Sigma \text{Sh } \lambda s \rightarrow \text{Po } s \rightarrow X$ 
open Con public
infixr 1  $\triangleleft$ 

```

Instead of having a *size* and a *vector* of contents, we represent the positions for each shape as a set, and the contents as a *function* from positions.

3.1 Closure Properties

We may readily check that the polynomials are all containers.

```

K $_{\triangleleft}$  : Set → Con
K $_{\triangleleft}$  A = A  $\triangleleft$   $\lambda _ \rightarrow \text{Zero}$ 

I $_{\triangleleft}$  : Con
I $_{\triangleleft}$  = One  $\triangleleft$   $\lambda _ \rightarrow \text{One}$ 

+ $_{\triangleleft}$  : Con → Con → Con
(S  $\triangleleft$  P) + $_{\triangleleft}$  (S'  $\triangleleft$  P') = (S + S')  $\triangleleft$   $\vee P \langle ? \rangle P'$ 

 $\times_{\triangleleft}$  : Con → Con → Con
(S  $\triangleleft$  P)  $\times_{\triangleleft}$  (S'  $\triangleleft$  P') = (S  $\times$  S')  $\triangleleft$   $\vee \lambda s s' \rightarrow P s + P' s'$ 

```

Moreover, we may readily close containers under dependent pairs and functions, a fact which immediately tells us how to compose containers.

```

 $\Sigma_{\triangleleft}$  : (A : Set) (C : A → Con) → Con
 $\Sigma_{\triangleleft}$  A C = ( $\Sigma A \lambda a \rightarrow \text{Sh } (C a)$ )  $\triangleleft$   $\vee \lambda a s \rightarrow \text{Po } (C a) s$ 

 $\Pi_{\triangleleft}$  : (A : Set) (C : A → Con) → Con
 $\Pi_{\triangleleft}$  A C = ((a : A) → Sh (C a))  $\triangleleft$   $\lambda f \rightarrow \Sigma A \lambda a \rightarrow \text{Po } (C a) (f a)$ 

 $\circ_{\triangleleft}$  : Con → Con → Con
(S  $\triangleleft$  P)  $\circ_{\triangleleft}$  C =  $\Sigma_{\triangleleft} S \lambda s \rightarrow \Pi_{\triangleleft} (P s) \lambda p \rightarrow C$ 

```

Exercise 3.1 (containers are endofunctors) Check that containers yield endofunctors which obey the laws.

```

conEndoFunctor : { C : Con } → EndoFunctor [ C ]Δ
conEndoFunctor { S Δ P } = ?
conEndoFunctorOKP : { C : Con } → EndoFunctorOKP [ C ]Δ
conEndoFunctorOKP { S Δ P } = ?

```

Exercise 3.2 (closure properties) Check that the meanings of the operations on containers are justified by their interpretations as functors.

3.2 Container Morphisms

A container morphism describes a *natural transformation* between the functors given by containers. As the element type is abstract, there is nowhere that the elements of the output can come from except somewhere in the input. Correspondingly, a container morphism is given by a pair of functions, the first mapping input shapes to output shapes, and the second mapping output positions back to the input positions from which they fetch elements.

```

→Δ : Con → Con → Set
(S Δ P) →Δ (S' Δ P') = Σ (S → S') λ f → (s : S) → P' (f s) → P s

```

The action of a container morphism is thus

```

/_{Δ} : forall { C C' } → C →Δ C' → forall { X } → [ C ]Δ X → [ C' ]Δ X
(to, fro) /_{Δ} (s, k) = to s, k ∘ fro s

```

Interactive Interpretation Peter Hancock encourages us to think of $S \Delta P$ as the description of a *command-response* protocol, where S is a set of commands we may invoke and P tells us which responses may be returned for each command. The type $[S \Delta P]_{\Delta} X$ is thus a *strategy* for obtaining an X by one run of the protocol. Meanwhile, a container morphism is thus a kind of ‘device driver’, translating commands one way, then responses the other.

Exercise 3.3 (representing natural transformations) Check that you can represent any natural transformation between containers as a container morphism.

```

morphΔ : forall { C C' } → (forall { X } → [ C ]Δ X → [ C' ]Δ X) → C →Δ C'
morphΔ f = ?

```

Container-of-positions presentation The above exercise might suggest an equivalent presentation of container morphisms, namely

$$(S \Delta P) \rightarrow_{\Delta} C = (s : S) \rightarrow [C]_{\Delta} (P s)$$

but the to-and-fro presentation is usually slightly easier to work with. You win some, you lose some.

Exercise 3.4 (identity and composition) Check that you can define identity and composition for container morphisms.

```

id→Δ : forall { C } → C →Δ C
id→Δ = ?
⊙→Δ : forall { C D E } → (D →Δ E) → (C →Δ D) → (C →Δ E)
e ⊙→Δ d = ?

```


3.3 W-types

The least fixpoint of a container is a \mathcal{W} -type— \mathcal{W} for ‘well founded’.

```
data  $\mathcal{W}$  (C : Con) : Set where
  ⟨-⟩ :  $\llbracket C \rrbracket_{\triangleleft} (\mathcal{W} C) \rightarrow \mathcal{W} C$ 
```

In an extensional setting, \mathcal{W} can be used to represent a great many datatypes, but intensional systems have some difficulties achieving faithful representations of first order data via \mathcal{W} -types.

Exercise 3.5 (natural numbers) Define natural numbers as a \mathcal{W} -type. Implement the constructors. Hint: *magic* : $\text{Zero} \rightarrow \{A : \text{Set}\} \rightarrow A$. Implement primitive recursion and use it to implement addition.

```
NatW : Set
NatW =  $\mathcal{W} ?$ 
zeroW : NatW
zeroW = ⟨?⟩
sucW : NatW → NatW
sucW n = ⟨?⟩
precW : forall {l} {T : Set l} → T → (NatW → T → T) → NatW → T
precW z s n = ?
addW : NatW → NatW → NatW
addW x y = precW ? ? x
```

How many different implementations of `zeroW` can you find? Meanwhile, discover for yourself why an attempt to establish the induction principle is a fool’s errand.

```
indW : forall {l} (P : NatW → Set l) →
  P zeroW →
  ((n : NatW) → P n → P (sucW n)) →
  (n : NatW) → P n
indW P z s n = ?
```

A useful deployment of the \mathcal{W} -type is to define the free monad for a container.

```
*- : Con → Set → Set
C * X =  $\mathcal{W} (K_{\triangleleft} X +_{\triangleleft} C)$ 
```

Exercise 3.6 (free monad) Construct the components for

```
freeMonad : (C : Con) → Monad (*- C)
freeMonad C = ?
```

Exercise 3.7 (free monad closure) Define an operator

```
_*_ $\triangleleft$  : Con → Con
*_ $\triangleleft$  C = ?
```

and exhibit an isomorphism

$$C * X \cong \llbracket C *_\triangleleft \rrbracket_{\triangleleft} X$$

Exercise 3.8 (general recursion) Define the monadic computation which performs one command-response interaction:

```
call : forall {C} → (s : Sh C) → C * Po C s
call s = ?
```

We can model, the general recursive function space as the means to perform finite, on in too much detail demand expansion of call trees.

```
Π⊥ : (S : Set) (T : S → Set) → Set
Π⊥ S T = (s : S) → (S < T) * T s
```

Give the ‘gasoline-driven’ interpreter for this function space, delivering a result provided the call tree does not expand more times than a given number.

```
gas : forall {S T} → ℕ → Π⊥ S T → (s : S) → T s + One
gas n f s = ?
```

Feel free to implement reduction for the untyped λ -calculus, or some other model of computation, as a recursive function in this way.

Turing completeness To say that Agda fails to be Turing complete is manifest nonsense. It does not stop you writing general recursive programs. It does not stop you feeding them to a client who is willing to risk running them. It does stop you giving a general recursive program a type which claims it is guaranteed to terminate, nor can you persuade Agda to execute such a program unboundedly in the course of checking a type. It is not unusual for typecheckers to refuse to run general recursive type-level programs. So the situation is *not* that we give up power for totality. Totality buys us a degree of honesty which partial languages just discard.

3.4 Derivatives of Containers

We have

$$[S < P]_{\triangleleft} X = \Sigma S \lambda s \rightarrow P s \rightarrow X$$

but we could translate the right-hand side into a more mathematical notation and observe that a container is something a bit like a power series:

$$[S < P]_{\triangleleft} X = \sum_{s:S} X^{(Ps)}$$

We might imagine computing a formal derivative of such a series, ‘multiplying down by each index, then subtracting one’, but we are not merely counting data—they have individual existences. Let us define a kind of ‘dependent decrement’, subtracting a *particular* element from a type.

```
-_ : (X : Set) (x : X) → Set
X - x = Σ X λ x' → x' ≃ x → Zero
```

That is, an element of $X - x$ is some element for X which is known to be other than x .

We may now define the formal derivative of a container.

```
∂ : Con → Con
∂ (S < P) = Σ S P < ∇ λ s p → P s - p
```

The shape of the derivative is the pair of a shape with one position, which we call the ‘hole’, and the positions in the derivative are ‘everywhere but the hole’.

Exercise 3.9 (plug) Exhibit a container morphism which witnesses the ability to fill the hole, provided equality on positions is decidable.

$$\begin{aligned} \text{plug} : \text{forall } \{C\} \rightarrow & ((s : \text{Sh } C) (p \ p' : \text{Po } C \ s) \rightarrow \text{Dec } (p \simeq p')) \rightarrow \\ & (\partial \ C \times_{\triangleleft} \text{I}_{\triangleleft}) \rightarrow_{\triangleleft} C \\ \text{plug } \{C\} \text{ } \text{poeq?} = & ? \end{aligned}$$

Exercise 3.10 (laws of calculus) Check that the following laws hold at the level of mutually inverse container morphisms.

$$\begin{aligned} \partial (K_{\triangleleft} A) &\cong K_{\triangleleft} \text{Zero} \\ \partial I &\cong K_{\triangleleft} \text{One} \\ \partial (C +_{\triangleleft} D) &\cong \partial C +_{\triangleleft} \partial D \\ \partial (C \times_{\triangleleft} D) &\cong (\partial C \times_{\triangleleft} D) +_{\triangleleft} (C \times_{\triangleleft} \partial D) \\ \partial (C \circ_{\triangleleft} D) &\cong (\partial C \circ_{\triangleleft} D) \times_{\triangleleft} \partial D \end{aligned}$$

What is $\partial (C *_{\triangleleft})$?

3.5 Denormalized Containers

These may appear later.

Chapter 4

Indexed Containers (Levitated)

There are lots of ways to present indexed containers, giving ample opportunities for exercises, but I shall use the Hancock presentation, as it has become my preferred version, too.

The idea is to describe functors between indexed families of sets.

```
record  $\triangleright$  (I J : Set) : Set1 where
  constructor  $\_ \triangleleft \_ \$ \_$ 
  field
    Sh : J → Set
    Po : (j : J) → Sh j → Set
    ri : (j : J) (s : Sh j) (p : Po j s) → I
     $\llbracket \_ \rrbracket_i$  : (I → Set) → (J → Set)
     $\llbracket \_ \rrbracket_i$  X j =  $\Sigma$  (Sh j)  $\lambda$  s → (p : Po j s) → X (ri j s p)
  open  $\triangleright$  public
```

An $I \triangleright J$ describes a J -indexed thing with places for I -indexed elements. Correspondingly, some $j : J$ tells us which sort of thing we're making, determining a shape set $\text{Sh } j$ and a position family $\text{Po } j$, just as with plain containers. The ri function then determines which I -index is demanded in each element position.

Interaction structures Hancock calls these indexed containers *interaction structures*. Consider J to be the set of possible 'states of the world' before an interaction, and I the possible states afterward. The 'before' states will determine a choice of commands we can issue, each of which has a set of possible responses which will then determine the state 'after'. An interaction structure thus describes the predicate transformer which describes the precondition for achieving a postcondition by one step of interaction. We are just using proof-relevant Hoare logic as the type system!

Exercise 4.1 (functoriality) We have given the interpretation of indexed containers as operations on indexed families of sets. Equip them with their functorial action for the following notion of morphism

```
 $\dot{\rightarrow} : \text{forall } \{I : \text{Set}\} \rightarrow (I \rightarrow \text{Set}) \rightarrow (I \rightarrow \text{Set}) \rightarrow \text{Set}$ 
 $X \dot{\rightarrow} Y = \text{forall } i \rightarrow X \ i \rightarrow Y \ i$ 
```

```
 $\text{ixMap} : \text{forall } \{I J\} \{C : I \triangleright J\} \{X Y\} \rightarrow (X \dot{\rightarrow} Y) \rightarrow \llbracket C \rrbracket_i X \dot{\rightarrow} \llbracket C \rrbracket_i Y$ 
 $\text{ixMap } f \ j \ xs = ?$ 
```

4.1 Petersson-Synek Trees

Kent Petersson and Dan Synek proposed a universal inductive family, amounting to the fixpoint of an indexed container

```
data Tree { J : Set } (C : J ▷ J) (j : J) : Set where
  ⟨_⟩ : [ C ]i (Tree C) j → Tree C j
```

The natural numbers are a friendly, if degenerate example.

```
NatC : One ▷ One
NatC = (λ _ → Two) ◁ (λ _ → Zero ⟨?⟩ One) $ _
zeroC : Tree NatC ⟨⟩
zeroC = ⟨ tt, magic ⟩
sucC : Tree NatC ⟨⟩ → Tree NatC ⟨⟩
sucC n = ⟨ ff, pure n ⟩
```

This is just the indexed version of the \mathcal{W} -type, so the same issue with extensionality arises.

We may also define the node structure for vectors as an instance.

```
VecC : Set → ℕ ▷ ℕ
VecC X = VS ◁ VP $ Vr where -- depending on the length
  VS : ℕ → Set
  VS zero = One -- nil is unlabelled
  VS (suc n) = X -- cons carried an element
  VP : (n : ℕ) → VS n → Set
  VP zero _ = Zero -- nil has no children
  VP (suc n) _ = One -- cons has one child
  Vr : (n : ℕ) (s : VS n) (p : VP n s) → ℕ
  Vr zero ⟨⟩ () -- nil has no children to index
  Vr (suc n) x ⟨⟩ = n -- the tail of a cons has the length one less
```

Let us at least confirm that we can rebuild the constructors.

```
vnil : forall {X} → Tree (VecC X) zero
vnil = ⟨ ⟨⟩, (λ ()) ⟩
vcons : forall {X n} → X → Tree (VecC X) n → Tree (VecC X) (suc n)
vcons x xs = ⟨ (x, (λ _ → xs)) ⟩
```

Why don't you have a go at rebuilding an inductive family in this manner?

Exercise 4.2 (simply typed λ -calculus) Define the simply typed λ -terms as Petersson-Synek trees.

```
STLC : (Cx ★ × ★) ▷ (Cx ★ × ★)
STLC = ?
```

Implement the constructors.

4.2 Closure Properties

It is not difficult to show that indexed containers have an identity composition which is compatible up to isomorphism with those of their interpretations.

Exercise 4.3 (identity and composition) *Construct*

$\text{IdIx} : \text{forall } \{I\} \rightarrow I \triangleright I$
 $\text{IdIx} = ?$

such that

$$\llbracket \text{IdIx} \rrbracket_i X\ i \cong X\ i$$

Similarly, construct the composition

$\text{Colx} : \text{forall } \{I\ J\ K\} \rightarrow J \triangleright K \rightarrow I \triangleright J \rightarrow I \triangleright K$
 $\text{Colx } C\ C' = ?$

such that

$$\llbracket \text{Colx } C\ C' \rrbracket_i X\ k \cong \llbracket C \rrbracket_i (\llbracket C' \rrbracket_i X)\ k$$

It may be useful to consider constructing binary products and coproducts, but let us chase after richer structure, exploiting dependent types to a greater extent. We may describe a class of indexed functors, as follows.

My motivation for level polymorphism will appear in due course.

data $\text{Desc } \{l\} (I : \text{Set } l) : \text{Set } (\text{Isuc } l)$ **where**
 $\text{var} : I \rightarrow \text{Desc } I$
 $\sigma \pi : (A : \text{Set } l) (D : A \rightarrow \text{Desc } I) \rightarrow \text{Desc } I$
 $\times_D : \text{Desc } I \rightarrow \text{Desc } I \rightarrow \text{Desc } I$
 $\kappa : \text{Set } l \rightarrow \text{Desc } I$
infixr 4 \times_D

which admit a direct interpretation as follows

$\llbracket - \rrbracket_D : \text{forall } \{l\} \{I : \text{Set } l\} \rightarrow \text{Desc } I \rightarrow (I \rightarrow \text{Set } l) \rightarrow \text{Set } l$
 $\llbracket \text{var } i \rrbracket_D X = X\ i$
 $\llbracket \sigma A D \rrbracket_D X = \sum A \lambda a \rightarrow \llbracket D\ a \rrbracket_D X$
 $\llbracket \pi A D \rrbracket_D X = (a : A) \rightarrow \llbracket D\ a \rrbracket_D X$
 $\llbracket D \times_D E \rrbracket_D X = \llbracket D \rrbracket_D X \times \llbracket E \rrbracket_D X$
 $\llbracket \kappa A \rrbracket_D X = A$

A family of such descriptions in $J \rightarrow \text{Desc } I$ thus determines, pointwise, a functor from $I \rightarrow \text{Set}$ to $J \rightarrow \text{Set}$. It is easy to see that every indexed container has a description.

$\text{ixConDesc} : \text{forall } \{I\ J\} \rightarrow I \triangleright J \rightarrow J \rightarrow \text{Desc } I$
 $\text{ixConDesc } (S \triangleleft P \S r) j = \sigma (S\ j) \lambda s \rightarrow \pi (P\ j\ s) \lambda p \rightarrow \text{var } (r\ j\ s\ p)$

Meanwhile, up to isomorphism at least, we can go the other way around.

Exercise 4.4 (from $J \rightarrow \text{Desc } I$ to $I \triangleright J$) *Construct functions*

$\text{DSh} : \{I : \text{Set}\} \rightarrow \text{Desc } I \rightarrow \text{Set}$
 $\text{DSh } D = ?$
 $\text{DPo} : \text{forall } \{I\} (D : \text{Desc } I) \rightarrow \text{DSh } D \rightarrow \text{Set}$
 $\text{DPo } D\ s = ?$
 $\text{Dri} : \text{forall } \{I\} (D : \text{Desc } I) (s : \text{DSh } D) \rightarrow \text{DPo } D\ s \rightarrow I$
 $\text{Dri } D\ s\ p = ?$

in order to compute the indexed container form of a family of descriptions.

$\text{desclxCon} : \text{forall } \{I\ J\} \rightarrow (J \rightarrow \text{Desc } I) \rightarrow I \triangleright J$
 $\text{desclxCon } F = (\text{DSh} \circ F) \triangleleft (\text{DPo} \circ F) \S (\text{Dri} \circ F)$

Exhibit the isomorphism

$$\llbracket \text{desclxCon } F \rrbracket_i X\ j \cong \llbracket F\ j \rrbracket_D X$$

We shall find further closure properties of indexed containers later, but let us explore description awhile.

4.3 Describing Datatypes

Descriptions are quite a lot like inductive family declarations. The traditional `Vec` declaration corresponds to

```
vecD : Set → ℕ → Desc ℕ
vecD X n =
  σ Two ( κ (n ≈ zero)
    ⟨?⟩ σ ℕ λ k → κ X ×D var k ×D κ (n ≈ suc k)
  )
```

The choice of constructors becomes a σ -description. Indices specialized in the return types of constructors become explicit equational constraints. However, in defining a family of descriptions, we are free to use the full computational power of the function space, inspecting the index, e.g.

```
vecD : Set → ℕ → Desc ℕ
vecD X zero = κ One
vecD X (suc n) = κ X ×D var n
```

To obtain a datatype from a description, we can turn it into a container and use the Petersson-Synek tree, or we can preserve the first orderness of first order things and use the direct interpretation.

```
data Data {l} {J : Set l} (F : J → Desc J) (j : J) : Set l where
  ⟨_⟩ : [ F j ]b (Data F) → Data F j
```

For example, let us once again construct vectors.

```
vnil : forall {X} → Data (vecD X) zero
vnil = ⟨ ⟩
vcons : forall {X n} → X → Data (vecD X) n → Data (vecD X) (suc n)
vcons x xs = ⟨ x, xs ⟩
```

Exercise 4.5 (something like ‘levitation’) Construct a family of descriptions which describes a type like `Desc`. As Agda is not natively cumulative, you will need to shunt types up through the `Set l` hierarchy by hand, with this gadget:

```
record ↑ {l} (X : Set l) : Set (lsuc l) where
  constructor ↑
  field
    ↓ : X
  open ↑
```

Now implement

```
DescD : forall {l} (I : Set l) → One {lsuc l} → Desc (One {lsuc l})
DescD {l} I _ = ?
```

Check that you can map your described descriptions back to descriptions.

```
desc : forall {l} {I : Set l} → Data (DescD I) ⟨ ⟩ → Desc I
desc D = ?
```

We could, if we choose, work entirely with described datatypes. Perhaps, in some future programming language, the external `Desc I` type will be identified with the internal `Data (DescD I) ⟨ ⟩` so that `Data` is the *only* datatype.

4.4 Some Useful Predicate Transformers

A container stores a bunch of data. If we have a predicate P on data, it might be useful to formulate the predicates on bunches of data asserting that P holds *everywhere* or *somewhere*. But an indexed container is a predicate transformer! We can thus close indexed containers under the formation of ‘everywhere’.

```

Everywhere : forall {I J} (C : I ▷ J) (X : I → Set) → Σ I X ▷ Σ J (⟦ C ⟧i; X)
Everywhere (S ◁ P $ r) X
  = (λ _ → One)
  ◁ (λ {(j, s, k)} → P j s)
  $ (λ {(j, s, k)} → p → r j s p, k p)

```

The witnesses to the property of the elements of the original container become the elements of the derived container. The trivial predicate holds everywhere.

```

allTrivial : forall {I J} (C : I ▷ J) (X : I → Set) jc →
  ⟦ Everywhere C X ⟧i (λ _ → One) jc
allTrivial C X _ = ⟨⟩, λ p → ⟨⟩

```

If you think of simply typed λ -calculus contexts as containers of types, then an *environment* is given by supplying values **Everywhere**.

Meanwhile, the finger now points at you, pointing a finger at an element.

Exercise 4.6 (Somewhere) Construct the transformer which takes C to the container for witnesses that a property holds for some element of a C -structure

```

Somewhere : forall {I J} (C : I ▷ J) (X : I → Set) → Σ I X ▷ Σ J (⟦ C ⟧i; X)
Somewhere (S ◁ P $ r) X
  = ?
  ◁ ?
  $ ?

```

Check that the impossible predicate cannot hold somewhere.

```

noMagic : forall {I J} (C : I ▷ J) (X : I → Set) jc →
  ⟦ Somewhere C X ⟧i (λ _ → Zero) jc → Zero
noMagic C X _ (p, m) = ?

```

For simply typed λ -calculus contexts, a variable of type T is just the evidence that a type equal to T is somewhere. Environment lookup is just the obvious property that if Q holds everywhere and R holds somewhere, then their conjunction holds somewhere, too.

Exercise 4.7 (lookup) Implement generalized environment lookup.

```

lookup : forall {I J} (C : I ▷ J) (X : I → Set) jc {Q R} →
  ⟦ Everywhere C X ⟧i Q jc → ⟦ Somewhere C X ⟧i R jc →
  ⟦ Somewhere C X ⟧i (λ ix → Q ix × R ix) jc
lookup C X jc qs r = ?

```

A key use of the **Everywhere** transformer is in the formulation of *induction* principles. The induction hypotheses amount to asserting that the induction predicate holds at every substructure..

```

treeInd : forall {I} (C : I ▷ I) (P : Σ I (Tree C) → Set) →
  ([ Everywhere C (Tree C) ]i P →
    (∀ λ i ts → P (i, ⟨ ts ⟩))) →
  (i : I) (t : Tree C i) → P (i, t)
treeInd C P m i ⟨ s, k ⟩ = m (i, s, k) (⟨ ⟩, λ p → treeInd C P m _ (k p))

```

The step method of the above looks a bit like an algebra, modulo plumbing.

Exercise 4.8 (induction as a fold) *Petersson-Synek trees come with a ‘fold’ operator, making `Tree C` (weakly) initial for `[C]i`. We can compute any P from a `Tree C`, given a C -algebra for P .*

```

treeFold : forall {I} (C : I ▷ I) (P : I → Set) →
  ([ C ]i P → P) →
  (Tree C → P)
treeFold C P m i ⟨ s, k ⟩ = m i (s, λ p → treeFold C P m _ (k p))

```

However, `treeFold` does not give us dependent induction on `Tree C`. If all you have is a hammer, everything looks like a nail. If we want to compute why some $P : \Sigma I (\text{Tree } C) \rightarrow \text{Set}$ always holds, we’ll need an indexed container storing P s in positions corresponding to the children of a given tree. The `Everywhere C` construct does most of the work, but you need a little adaptor to unwrap the C container inside the `Tree C`.

```

Children : forall {I} (C : I ▷ I) → Σ I (Tree C) ▷ Σ I (Tree C)
Children C = Colx ? (Everywhere C (Tree C))

```

Now, you can extract a general induction principle for `Tree C` from `treeFold (Children C)`, but you will need a little construction. Finish the job.

```

treeFoldInd : forall {I} (C : I ▷ I) P →
  ([ Children C ]i P → P) →
  forall it → P it
treeFoldInd C P m (i, t) = treeFold (Children C) P m (i, t) ?

```

Of course, you need to do what is effectively an inductive proof to fill in the hole. Induction really does amount to more than weak initiality. But one last induction will serve for all.

What goes for containers goes for descriptions. We can build all the equipment of this section for `Desc` and `Data`, too.

Exercise 4.9 (Everywhere and Somewhere for Desc) *Define suitable description transformers, capturing what it means for a predicate to hold in every or some element position within a given described structure.*

```

EverywhereD SomewhereD : { I : Set } (D : Desc I) (X : I → Set) →
  [ D ]b X → Desc (Σ I X)
EverywhereD D X xs = ?
SomewhereD D X xs = ?

```

Now construct

```

dataInd : forall { I : Set } (F : I → Desc I) (P : Σ I (Data F) → Set) →
  ((i : I) (ds : [ F i ]b (Data F)) →
    [ EverywhereD (F i) (Data F) ds ]b P → P (i, ⟨ ds ⟩)) →
  forall i d → P (i, d)
dataInd F P m i d = ?

```

4.5 Indexed Containers are Closed Under Fixpoints

So far, we have used indexed containers to describe the node structures of recursive data, but we have not considered recursive data structures to be containers themselves. Consider, e.g., the humble vector: might we not consider the vector's elements to be a kind of contained thing, just as much as its subvectors? We can just throw in an extra kind of element!

```
vecNodeIx : (One + ℕ) ▷ ℕ
vecNodeIx = descIxCon {J = ℕ} λ
  { zero    → κ One
  ; (suc n) → var (tt, ⟨⟩) ×D var (ff, n)
  }
```

That is enough to see vector *nodes* as containers of elements or subnodes, but it still does not give *vectors* as containers:

```
vecIx : One ▷ ℕ
vecIx = ?
```

We should be able to solve this goal by taking `vecNodeIx` and tying a recursive knot at positions labelled (ff, n) , retaining positions labelled $(\text{tt}, \langle \rangle)$. Let us try the general case.

```
μIx : forall {I J} → (I + J) ▷ J → I ▷ J
μIx {I} {J} F = (Tree F' ◦ -, - ff) ◁ (P' ◦ -, - ff) § (r' ◦ -, - ff) where
```

The shapes of the recursive structures are themselves trees, with unlabelled leaves at I -indexed places and F -nodes in J -indexed places. We could try to work in $J ▷ J$, cutting out the non-recursive positions. However, it is easier to shift to $(I + J) ▷ (I + J)$, introducing ‘unlabelled leaf’ as the dull node structure whenever an I shape is requested. We may construct

```
F : (I + J) ▷ (I + J)
F = (v (λ i → One)    ⟨?⟩ Sh F)
  ◁ (v (λ _ → Zero)   ⟨?⟩ Po F)
  § (v (λ t s ())     ⟨?⟩ ri F)
```

and then choose to start with (ff, j) for the given top level j index. A position is then a path to a leaf: either we are at a leaf already, or we must descend further.

```
P : (x : I + J) → Tree F x → Set
P (tt, i) _      = One
P (ff, j) ⟨ s, k ⟩ = Σ (Po F j s) λ p → P (ri F j s p) (k p)
```

Finally, we may follow each path to its indicated leaf and return the index which sent us there.

```
r : (x : I + J) (t : Tree F x) → P x t → I
r (tt, i) _      = i
r (ff, j) ⟨ s, k ⟩ (p, ps) = r _ (k p) ps
```

Let us check that this recipe cooks the vectors.

```
vecIx : One ▷ ℕ
vecIx = μIx vecNodeIx
```

```

Vec : Set → ℕ → Set
Vec X =  $\llbracket \text{vec} \rrbracket_i (\lambda \_ \rightarrow X)$ 
vnil : forall {X} → Vec X zero
vnil =  $\langle \langle \rangle, \lambda () \rangle, (\vee \lambda ())$ 
vcons : forall {X n} → X → Vec X n → Vec X (suc n)
vcons x (s, k)
  =  $\langle \_, (\lambda \{(\text{tt}, \_) \rightarrow \langle \_, \lambda () \rangle; (\text{ff}, \_) \rightarrow s\}) \rangle$ 
  ,  $(\lambda \{((\text{tt}, \_), \_) \rightarrow x; ((\text{ff}, \_), p) \rightarrow k\} p)$ 

```

4.6 Adding fixpoints to Desc

We can extend descriptions to include a fixpoint operator:

```

data Desc (I : Set) : Set1 where
  var : I → Desc I
  σ π : (A : Set) (D : A → Desc I) → Desc I
  ×D : Desc I → Desc I → Desc I
  κ : Set → Desc I
  μ : (J : Set) → (J → Desc (I + J)) → J → Desc I

```

The interpretation must now be defined mutuallu with the universal inductive type.

```

mutual
   $\llbracket \_ \rrbracket_{\text{D}}$  : forall {I} → Desc I → (I → Set) → Set
   $\llbracket \text{var } i \rrbracket_{\text{D}}$  X = X i
   $\llbracket \sigma A D \rrbracket_{\text{D}}$  X =  $\Sigma A \lambda a \rightarrow \llbracket D a \rrbracket_{\text{D}}$  X
   $\llbracket \pi A D \rrbracket_{\text{D}}$  X = (a : A) →  $\llbracket D a \rrbracket_{\text{D}}$  X
   $\llbracket D \times_D E \rrbracket_{\text{D}}$  X =  $\llbracket D \rrbracket_{\text{D}}$  X ×  $\llbracket E \rrbracket_{\text{D}}$  X
   $\llbracket \kappa A \rrbracket_{\text{D}}$  X = A
   $\llbracket \mu J F j \rrbracket_{\text{D}}$  X = Data F X j
  data Data {I J} (F : J → Desc (I + J)) (X : I → Set) (j : J) : Set where
     $\langle \_ \rangle$  :  $\llbracket F j \rrbracket_{\text{D}}$  (v X  $\langle ? \rangle$  Data F X) → Data F X j

```

Indeed, `Desc Zero` now does quite a good job of reflecting `Set`, except that the domains of `σ` and `π` are not concretely represented, an issue we shall attend to in the next chapter.

Exercise 4.10 (induction) *State and prove the induction principle for Desc. (This is not an easy exercise.)*

4.7 Jacobians

I am always amused when computing people complain about being made to learn mathematics choose calculus as their favourite example of something that is of no use to them. I, for one, am profoundly grateful to have learned vector calculus: it is exactly what you need to develop notions of ‘context’ for dependent datatypes.

An indexed container in $I \triangleright J$ explains J sorts of structure in terms of I sorts of elements, and as such, we acquire a *Jacobian matrix* of partial derivatives, in $I \triangleright (J \times I)$. A (j, i) derivative is a structure of index j with a hole of index i . Here’s how we build it.

$$\begin{aligned}
\mathcal{J} &: \text{forall } \{I J\} \rightarrow I \triangleright J \rightarrow I \triangleright (J \times I) \\
\mathcal{J} (S \triangleleft P \S r) &= (\lambda \{ (j, i) \rightarrow \Sigma (S j) \lambda s \rightarrow r j s^{-1} i \}) \\
\triangleleft & (\lambda \{ (j, \cdot (r j s p)) (s, \text{from } p) \rightarrow P j s - p \}) \\
\S & (\lambda \{ (j, \cdot (r j s p)) (s, \text{from } p) (p', -) \rightarrow r j s p' \})
\end{aligned}$$

The shape of an (i, j) -derivative must select a j -indexed shape for the structure, together with a position (the hole) whose index is i . As in the simple case, a position in the derivative is any position other than the hole, and its index is calculated as before.

Exercise 4.11 (plugging) Check that a decidable equality for positions is enough to define the ‘plugging in’ function.

Einstein’s summation convention might be useful to infer the choice and placement of quantifiers.

$$\begin{aligned}
\text{plug} &: \text{forall } \{I J\} (C : I \triangleright J) \rightarrow \\
& ((j : J) (s : \text{Sh } C j) (p p' : \text{Po } C j s) \rightarrow \text{Dec } (p \simeq p')) \rightarrow \\
& \text{forall } \{i j X\} \rightarrow \llbracket \mathcal{J} C \rrbracket_i X (j, i) \rightarrow X i \rightarrow \llbracket C \rrbracket_i X j \\
\text{plug } C \text{ eq? } jx x &= ?
\end{aligned}$$

Exercise 4.12 (the Zipper) For a given $C : I \triangleright I$, construct the indexed container $\text{Zipper } C : (I \times I) \triangleright (I \times I)$ such that $\text{Tree } (\text{Zipper } C) (ir, ih)$ represents a one ih -hole context in a $\text{Tree } C$ ir , represented as a sequence of hole-to-root layers.

$$\begin{aligned}
\text{Zipper} &: \text{forall } \{I\} \rightarrow I \triangleright I \rightarrow (I \times I) \triangleright (I \times I) \\
\text{Zipper } C &= ?
\end{aligned}$$

Check that you can zipper all the way out to the root.

$$\begin{aligned}
\text{zipOut} &: \text{forall } \{I\} (C : I \triangleright I) \{ir ih\} \rightarrow \\
& ((i : I) (s : \text{Sh } C i) (p p' : \text{Po } C i s) \rightarrow \text{Dec } (p \simeq p')) \rightarrow \\
& \text{Tree } (\text{Zipper } C) (ir, ih) \rightarrow \text{Tree } C ih \rightarrow \text{Tree } C ir \\
\text{zipOut } C \text{ eq? } cz t &= ?
\end{aligned}$$

Exercise 4.13 (differentiating Desc) The notion corresponding to \mathcal{J} for descriptions is ∇ , computing a ‘vector’ of partial derivatives. Define it symbolically.

“grad”

$$\begin{aligned}
\nabla &: \{I : \text{Set}\} \rightarrow \text{Desc } I \rightarrow I \rightarrow \text{Desc } I \\
\nabla D h &= ?
\end{aligned}$$

Symbolic differentiation is the first example of a pattern matching program in my father’s thesis (1970).

Hence construct suitable zippering equipment for **Data**.

It is amusing to note that the mathematical notion of *divergence*, $\nabla \cdot D$, corresponds exactly to the choice of decompositions of a D -structure into any element-in-context:

$$\sigma I \lambda i \rightarrow \nabla D i \times_D \text{var } i$$

I have not yet found a meaning for *curl*, $\nabla \times D$, nor am I expecting Maxwell’s equations to pop up anytime soon. But I live in hope for light.

Chapter 5

Induction-Recursion

Chapter 6

Observational Equality

Chapter 7

Type Theory in Type Theory

Chapter 8

Reflections and Directions

Bibliography

Rod Burstall. Proving properties of programs by structural induction. *Computer Journal*, 12(1):41–48, 1969.

Ulf Norell. Dependently typed programming in agda. In Pieter W. M. Koopman, Rinus Plasmeijer, and S. Doaitse Swierstra, editors, *Advanced Functional Programming*, volume 5832 of *LNCS*, pages 230–266. Springer, 2008.